
Manual de cálculo científico en SageMath

¿Qué es SageMath?

SageMath, conocido anteriormente como Sage, es un sistema algebraico computacional que destaca por
tres características:

Estar construido sobre paquetes matemáticos ya contrastados como NumPy, Sympy, PARI/GP o Ma-
xima.

Acceder a la potencia combinada de los mismos a través de un lenguaje común basado en Python.

SageMath es un software libre.

La interacción con el usuario es posible desde la interfaz de línea de comandos (terminal interactiva)
o vía Notebook que combinan celdas de código con celdas con gráficos, texto enriquecido o fórmulas
con LATEX. Este software matemático permite hacer varios cálculos en áreas como Álgebra, Combinatoria,
Cálculos numéricos, Cálculo, etc, además es una herramienta gráfica versátil.

SageManifolds implementa Geometría Diferencial y Cálculo Tensorial en SageMath, herramientas esen-
ciales para cálculos en Relatividad General y demás áreas científicas.

1. Instalación

1.1. Instalación de SageMath
Para instalar SageMath se debe tener una computadora con al menos 2 GB de espacio libre en el disco

duro y con sistema operativo Linux (32-bit o 64-bit) u OS X (10.4 o posterior). Para instalarlo en Windows
se necesita una máquina virtual. Se recomienda tener instalado LATEX.

1.1.1. Linux

Para instalar SageMath en Linux seguimos los siguientes pasos:

1. Nos dirigimos a la dirección web:

http://www.sagemath.org/

2. Damos click en Download 6.7. La versión 6.7 de SageMath es la más reciente al 15 de junio del
2015.

3. Aparecerá una lista de servidores para descargar el binario para Linux. Escogemos uno dando click
sobre el nombre. Por ejemplo, damos click en MIT, Cambridge, MA, USA.

4. Se mostrará una pantalla donde debemos escoger el número de bits de nuestra computadora y dar
click sobre 32bit o 64bit. Para saber de cuantos bits es nuestra computadora vamos a
Configuracion del sistema luego en Detalles y en Tipo de SO estará el número de
bits.

5. Una vez que se da click en el número de bits, aparecerá en otra pantalla una lista de archivos. Damos
click sobre el archivo dependiendo del sistema operativo. Por ejemplo, para Ubuntu 14.04 damos
click sobre
sage-6.7-x86_64-Linux-Ubuntu_14.04_x86_64.tar.gz
y luego sobre Guardar archivo para descargarlo.

6. Una vez finalizada la descarga copiamos el archivo al escritorio.

7. Debemos descomprimir el archivo, lo cual lo podemos hacer desde la terminal. Para esto primero
abrimos el terminal (Ctrl+Alt+T), y nos dirigimos al directorio donde esta el archivo descargado,
es decir, al Escritorio con el siguiente comando

cd Escritorio

o con

cd Desktop

dependiendo del idioma. El comando cd sirve para poder cambiar de directorios-carpetas. Una vez
que hemos cambiado de directorio, dependiendo del tipo de archivo descargado ponemos el siguiente
comando para descomprimir:

Para archivos .tar.gz:
tar -xzvf *.tar.gz

Para archivos .tar.lrz:
sudo apt-get install lrzip

lrzuntar *.tar.lrz

En lugar de * va el nombre del archivo descargado.

8. Se creará una carpeta en el escritorio. Por ejemplo, sage-6.7-x86_64-Linux. Cerramos la ter-
minal.

9. Para ejecutar SageMath lo hacemos desde la terminal (Ctrl+Alt+T). Cambiamos de directorio con
el comando cd hasta el directorio de SageMath. Por ejemplo, si la carpeta descomprimida se llama
sage-6.7-x86_64-Linux y está en el escritorio ponemos el comando

cd Escritorio/sage-6.7-x86_64-Linux

que es el directorio de Sage y luego el comando ./sage para ejecutarlo.

1.1.2. Windows

Para instalar SageMath en Windows seguimos los siguientes pasos:

1. Primero debemos instalar VirtualBox. Vamos a

https://www.virtualbox.org/

2. Damos click en Downloads (Descargas)

3. Damos click en

VirtualBox 4.3.28 for Windows hosts x86/amd64

y esperamos que se descargue el archivo.

4. Ejecutamos el archivo descargado VirtualBox-4.3.28-100309-Win.exe.

5. Damos click en Next (Siguiente) dejando los valores predeterminados, hasta que aparezca un men-
saje de
Warning: Network Interfaces y le damos click a Yes (si) para permitir la instalación.

6. Le damos click en Instalar.

7. Aparecerán mensajes diciendo:

Desea instalar este software de dispositivo?

y le damos click en Instalar en cada mensaje.

8. Una vez concluida la instalación le damos click a Finish (Finalizar).

9. Después se debe instalar la máquina virtual de SageMath. Para esto vamos a

http://www.sagemath.org/download-windows.html

10. Aparecerá una lista de servidores de descarga. Deberemos escoger uno y dar click sobre su nombre.
Por ejemplo,

Universite Pierre et Marie Curie, Paris, France.

11. Se mostrará una lista de archivos. Le damos click a sage-6.7.ova para descargarlo. La versión
6.7 es la más reciente de SageMath al 15 de junio del 2015.

12. Una vez descargado el archivo, ejecutamos Oracle VM VirtualBox.

13. En el menú Archivo, damos click en Importar servicio virtualizado (Ctrl+I).

14. Direccionamos al importador hasta donde se encuentra el archivo descargado sage-6.7.ova y
damos click en Next (Siguiente).

15. Damos click en Importar. Si se tienen problemas con la importación de la máquina virtual se
debe descargar desde otro servidor el archivo sage-6.7.ova. Los archivos descargados desde el
servidor

Universite Pierre et Marie Curie, Paris, France

y desde el servidor MIT, Cambridge, MA, USA han sido probados y funcionan.

16. Una vez finalizada la importación, en Oracle VM VirtualBox aparecerá en la lista de máquinas
virtuales la máquina Sage-6.7. Seleccionamos la máquina virtual y damos click en Iniciar y
esperamos que se inicie. Aparecerá en pantalla la interfaz The Notebook de SageMath.

17. Para iniciar SageMath se lo debe hacer desde la máquina virtual repitiendo el paso anterior.

1.1.3. Mac OS X

La instalación en Mac OS X se la realiza como una aplicación.

1. Nos dirigimos a

http://mirrors.mit.edu/sage/osx/index.html

Si se tiene una versión anterior a la versión 10.10, se puede dar click en intel y se mostrará una
lista de más archivos para descargar dependiendo de la versión de Mac OS X. Damos click sobre el
nombre del archivo cuyo nombre termine en app.dmg para descargar el binario, por ejemplo
sage-6.6-x86_64-Darwin-OSX_10.7_x86_64-app.dmg
para la versión 10.7. Se puede probar con otros servidores de descarga en

http://www.sagemath.org/download-mac.html

2. Una vez descargado el archivo damos doble click sobre el mismo.

3. Se abrirá una unidad de disco virtual, dentro de la cual estará un archivo .txt y otro con el logo de
Sage. A este último lo arrastramos hasta Aplicaciones.

4. Desmontamos el disco virtual.

5. En Aplicaciones damos doble click sobre la aplicación de Sage.

6. En el explorador se abrirá The Sage Notebook mientras que para abrir desde el terminal damos click
en el icono de Sage que aparecerá en la barra superior y le damos click en Terminal Session y
luego en Sage.

1.2. Instalación de SageManifolds
Previo a instalar SageManifolds se deben tener instalados Sage (6.0 o superior) y Python (2.7 o superior).

Para la instalación en Mac, adicionalmente se debe tener instalado GNU sed.

1.2.1. Linux

1. Ir a

http://sagemanifolds.obspm.fr/spkg/sm-install.sh

2. Descargar el archivo sm-install.sh.

3. Copiar el archivo al directorio de Sage, es decir a la carpeta desde donde se ejecuta SageMath (carpeta
donde se descomprimió el archivo descargado para instalar SageMath)

4. Abrir el terminal (Ctrl+Alt+T).

5. Cambiar al directorio de SageMath y ejecutar el comando

bash sm-install.sh

o

sudo bash sm-install.sh

6. Una vez finalizada la instalación se puede ejecutar SageMath (./sage desde el directorio de Sage-
Math) y hacer la siguiente prueba: Escribir en el terminal el siguiente comando

sage: Manifold(2, ’M’)

y si se imprime en pantalla

Manifold(2, ’M’)

la instalación de SageManifolds está correcta.

1.2.2. Windows

1. Iniciar SageMath desde la máquina virtual.

2. Una vez iniciado Sage, oprimir Ctrl(derecha)+F1 hasta que aparezca la terminal (pantalla ne-
gra)

3. Desde la terminal ingresamos lo siguiente:

login: sage

password: sage

4. Escribimos el siguiente comando:

sage -root

y nos indicará el directorio root de Sage, por ejemplo /home/sage/sage-6.7.

5. Cambiamos al directorio root de Sage con el comando

cd /home/sage/sage-6.3

6. Escribimos los siguientes comandos para la instalación de SageManifolds (se debe tener conexión a
internet):

wget http://sagemanifolds.obspm.fr/spkg/sm-install.sh

bash sm-install.sh

7. Cerramos la máquina virtual.

8. Al ejecutar nuevamente SageMath desde la máquina virtual se puede comprobar que se encuentra ins-
talado SageManifolds al escribir en una de las celdas de programación de la Notebook de SageMath
el siguiente comando

sage: Manifold(2, ’M’)

y si se imprime en pantalla

Manifold(2, ’M’)

la instalación de SageManifolds está correcta.

1.2.3. Mac OS X

1. Descargamos de

http://sagemanifolds.obspm.fr/download.html

el archivo sm-install-macos-bsd.sh que se encuentra en la sección de Mac Os X en el link
anterior.

2. Ejecutamos una sesión de Sage en el terminal (click en la aplicación de Sage, luego damos click en
el icono de Sage en la barra superior, click en Terminal Session y luego en Sage).

3. En el terminal aparecerá el directorio root de Sage del tipo

/Applications/Sage-6.6.app/Contents/Resources/sage

Copiamos el archivo descargado al directorio root de Sage. Desde el terminal lo podemos hacer de
la siguiente manera: Con el comando cd nos dirigimos al directorio donde se encuentra el archivo
descargado y con la siguiente línea de comando lo copiamos al directorio de Sage

cp sm-install-macos-bsd.sh /Applications/Sage-6.6.app/Contents/Resources/sage

donde se tiene el nombre del archivo descargado y el directorio de Sage como ejemplo.

4. Después desde el terminal cambiamos de directorio hasta el directorio de Sage y ponemos el siguiente
comando

bash sm-install-macos-bsd.sh

y se completará la instalación.

5. Para comprobar la instalación ejecutamos Sage e iniciamos una sesión desde el terminal. En el termi-
nal escribimos

sage: Manifold(2, ’M’)

y si se imprime en pantalla

Manifold(2, ’M’)

la instalación de SageManifolds está correcta.

2. SageMath

2.1. La terminal interactiva
2.1.1. Inicio de sesión

Para entrar a la interfaz interactiva de SageMath lo hacemos dependiendo del Sistema Operativo de
nuestra computadora.

En Linux, ingresamos al terminal y lo direccionamos al directorio de Sage con el comando cd. Una
vez en el directorio, escribimos el comando ./sage y por defecto se abre la terminal interactiva. En
la figura (1) se muestra la terminal interactiva en Linux.

En Windows, ejecutamos la máquina virtual de Sage desde VirtualBox. Una vez abierta la inter-
faz The Notebook de SageMath oprimimos Ctrl(derecha)+F1 hasta que aparezca el terminal
(pantalla negra). En la terminal ingresamos los siguiente:

login: sage

password: sage
Escribimos el comando sage y se abrirá la terminal interactiva.

En Mac OS X, ejecutamos una sesión de Sage en la terminal dando click en la aplicación de Sage,
luego damos click en el icono de Sage en la barra superior, click en Terminal Session y luego
en Sage. Con esto se abrirá la terminal inteactiva.

Figura 1: Inicio de sesión en la terminal interactiva de SageMath

2.1.2. Fin de sesión

Para finalizar la sesión se tienen tres opciones:

Escribimos el comando exit

Escribimos el comando quit

Oprimimos Ctrl+D

En la figura (2) se muestra la terminal interactiva de Sage en Linux luego de ejecutar el comando quit.

Figura 2: Fin de sesión en terminal interactiva de SageMath

2.1.3. Comandos de la terminal de UNIX

Cualquier comando de UNIX puede ser ejecutado desde la terminal interactiva de SageMath si se ante-
pone el símbolo ! antes del comando, por ejemplo:

sage: !ls

para mostrar los archivos dentro del directorio. Además, la terminal interactiva ignora los prompt, es decir

sage: sage: 5+3
8

es lo mismo que

sage: sage: sage: 5+3
8

y que

sage: >>> 5+3
8

El hecho de que SageMath ignore los prompt es útil en caso de que se copien comandos de otras sesiones.

2.1.4. Sage como calculadora

Desde la terminal interactiva se pueden hacer operaciones propias de una calculadora como

Suma: la operación se realiza con el símbolo +, a continuación un ejemplo

sage: 5+2
7

Resta: la operación se realiza con el símbolo -, a continuación un ejemplo

sage: 5-2
3

Multiplicación: la operación se realiza con el símbolo *, a continuación un ejemplo

sage: 5*2
10

División: la operación se realiza con el símbolo /, a continuación un ejemplo

sage: 5/2
5/2

Potenciación: la operación se realiza con el símbolo ^, a continuación un ejemplo

sage: 5^2
25

Raíz cuadrada: la operación se realiza con la función sqrt(), a continuación un ejemplo

sage: sqrt(4)
2

Si queremos que en pantalla se muestre, por ejemplo, el resultado de 7/3 con decimales debemos usar
cualquiera de las siguientes líneas de comando

1. sage: float(7/3)
2.33333333333333

2. sage: 7./3
2.33333333333333

3. sage: 7./3.
2.33333333333333

Se pueden usar los mismos comandos con cualquier otra operación para mostrar en pantalla el resultado
con decimales. También se puede usar notación científica. Por ejemplo,

sage: 1+10^(-1)
11/10
sage: float(1+10^(-1))
1.1

donde la primera línea de comando muestra en pantalla el resultado de la operación sin decimales y la
segunda línea el resultado con decimales.

2.1.5. Navegar en la sesión

Se tiene una lista de comandos que permiten mostrar las entradas y las salidas en la sesión:

El comando sage: _ muestra en pantalla la última salida.

El comando sage: __ muestra en pantalla la penúltima salida.

El comando sage: _oh muestra en pantalla una lista de todas las salidas numeradas desde la pri-
mera salida hasta la última.

El comando sage: _k muestra en pantalla la salida k-ésima. Las salidas están numeradas desde
el uno.

El comando sage: _ik muestra en pantalla la entrada k-ésima. Las entradas están numeradas
desde el uno.

El comando sage: %hist muestra en pantalla el historial de entradas.

Por ejemplo si luego de iniciar sesión en la terminal interactiva escribimos los comandos

sage: 5+8
13
sage: 7*3
21

y queremos que se muestre en pantalla la penúltima salida debemos poner

sage: __
13

En este caso como la primera y penúltima salida coinciden, podemos usar también el comando

sage: _1
13

Si deseamos ver la primera entrada, colocamos el comando

sage: _i1
u’5+8’

Con el comando eval() podemos evaluar una expresión. Por ejemplo, si queremos evaluar la primera
entrada escribiremos

sage: eval(_i1)
13

Para mostrar el historial de salidas usamos

sage: _oh
{1: 13, 2: 21, 3: 13,
4: 13, 5: u’5+8’, 6: 13}

y para el historial de entradas

sage: %hist
5+8
7*3
__
_1
_i1
eval(_i1)
_oh
%hist

2.1.6. Macros

Las macros permiten ejecutar una serie de tareas y son de gran utilidad cuando se deben realizar tareas
repetitivas. En las macros se guardan las tareas repetitivas y para ejecutarlas a todas se debe ejecutar la
macro únicamente.

A continuación mostramos la creación de una macro desde la terminal interactiva de SageMath con un
ejemplo: Ingresamos los comandos

sage: a = 5
sage: b = 2

Para mostrar los valores de a y b usamos el comando

sage: a,b
(5, 2)

Si en cada sesión debemos ingresar que a = 5 y b = 2, es mejor guardar este proceso en una macro.
Para crearla usamos el siguiente comando

sage: %macro miMacro 1 2
Macro ‘miMacro‘ created. To execute, type its name (without quotes).
=== Macro contents: ===
a = Integer(5)
b = Integer(2)

donde miMacro es el nombre de la macro y 1 2 son los números de las entradas que estamos guardando
para que se ejecuten con la macro. Para ejecutar la macro lo hacemos con su nombre, en este caso con el
comando sage: miMacro. Si por ejemplo, cambiamos el valor de a

sage: a = 3
sage: a,b
(3, 2)

y luego ejecutamos la macro miMacro

sage: miMacro

entonces, el valor de a volverá a ser el mismo ingresado en la entrada 1. Lo podemos comprobar mostrando
los valores de a y b al evaluar

sage: a,b
(5, 2)

2.1.7. Cronometrar

Al usar el comando sage: %time al principio de alguna línea de entrada, se muestra en pantalla el
tiempo que se demora el programa en ejecutar dicha línea y después muestra el resultado del comando
ingresado.

Por ejemplo, usando el lenguaje nativo de Python, se puede establecer el tiempo que se demora la
computadora en asignar a una variable a el valor 193899484. Para eso usamos la siguiente línea de comando

sage: %time a = int(1938)^int(99484)
CPU times: user 944 ms, sys: 4 ms, total: 948 ms
Wall time: 953 ms

El CPU times es el tiempo que el procesador estuvo activo realizando la operación, mientras que el
Wall time es el tiempo que transcurrió desde que inició la operación hasta que finalizó. También se puede
usar el lenguaje nativo de Sage de la siguiente forma

sage: %time a = 1938^99484
CPU times: user 52 ms, sys: 0 ns, total: 52 ms
Wall time: 50.5 ms

También se puede usar la interfaz de la libreria C PARI:

sage: %time a = pari(1938)^pari(99484)
CPU times: user 56 ms, sys: 4 ms, total: 60 ms
Wall time: 134 ms

El tiempo variará dependiendo de la computadora. Además se puede establecer el tiempo que le toma
a la computadora ejecutar una serie de comandos usando sage: cputime() y sage: walltime(),
los cuales deben asignarse a una variable. Por ejemplo, con las siguientes líneas de entradas

sage: t = cputime()
sage: a = int(1938)^int(99484)
sage: b = 1938^99484
sage: c = pari(1938)^pari(99484)
sage: cputime(t)
1.208000000000002

se muestra el tiempo que estuvo activo el procesador para asignar los valores a las variables a, b y c. Si se
quiere conocer el tiempo que trascurrió desde que empezó la asignación hasta que terminó se usan las líneas
siguientes

sage: t = walltime()
sage: a = int(1938)^int(99484)
sage: b = 1938^99484
sage: c = pari(1938)^pari(99484)
sage: walltime(t)
17.64968514442444

2.1.8. Errores

Cuando hay algún tipo de error en la línea de entrada, un mensaje de error se muestra en pantalla después
de ejecutar la línea. El mensaje de error sugerirá el tipo de error y donde se encuentra posiblemente. Por
ejemplo si ingresamos la siguiente línea

sage: 3_2
File "<ipython-input-4-8385810722b7>", line 1

3_2
^

SyntaxError: invalid syntax

se muestra un error de sintaxis dado que _ no representa ninguna operación entre 3 y 2, así también den-
tro del mensaje se indica donde está el error. Es posible que se produzcan errores por usar variables no
declaradas, por ejemplo si ponemos la siguiente línea sin antes declarar a a como variable

sage: a
---------------------------------------------------------------------------
NameError Traceback (most recent call last)
<ipython-input-6-60b725f10c9c> in <module>()
----> 1 a

NameError: name ’a’ is not defined

se tendrá un error y el mensaje nos sugiere que es porque a no ha sido declarada.

2.1.9. Búsqueda, inversa y autocompletar

Para realizar una búsqueda sobre cada entrada que se haya realizado se puede oprimir Ctrl+p o la
flecha arriba para navegar por la entradas. La búsqueda empezará desde la última entrada y seguirá en
orden hasta la primera entrada. La búsqueda se extenderá hasta entradas realizadas en sesiones anteriores.

Para realizar una bśqueda en el historial de entradas oprimimos Ctrl+r y escribimos las letras iniciales
de la entrada que estamos buscando.

La tecla Tab se utiliza para autocompletar. A continuación un ejemplo,

sage: V=VectorSpace(QQ,3)
sage: V
Vector space of dimension 3 over Rational Field
sage: V.z[Tab]
V.zero V.zero_submodule V.zero_vector
V.zero_element V.zero_subspace
sage: V.zero

En la primera línea se crea el objeto V y en la segunda se muestra el objeto. En la tercera línea se oprime
Tab luego de escribir V.z y en pantalla se muestran todas las funciones asociadas al objeto cuya letra
inicial es z. Si existiese una única opción luego de oprimir Tab, el texto se autocompleta.

2.1.10. Ayuda

Para acceder a la ayuda de los comandos, se usan las siguientes líneas de entrada:

Nombre? se muestra ayuda del comando Nombre, la cual incluye los objetos que requiere de entrada
el comando e incluye ejemplos. Se debe oprimir la flecha abajo para seguir leyendo la ayuda. Para
salir escribimos q.

Nombre?? se muestra el código fuente del comando Nombre. También incluye ejemplos y los
objetos de entrada. Para salir escribimos q.

help() se muestra documentación de algún módulo. Dentro de los paréntesis se escribe el nombre
del módulo.

Por ejemplo, con el objeto V creado en la sección anterior podemos obtener información de su función
llamada coordinates con los siguientes comandos

sage: V.coordinates?
sage: V.coordinates??
sage: help(VectorSpace)

Con la última entrada se obtiene la documentación de los objetos VectorSpace.

2.1.11. Guardar y cargar

Para guardar objetos individuales se usa el comando save(A,’A’) donde A es el nombre del objeto
que se está guardando y ’A’ es el nombre con el cual se lo guarda. Para cargar el objeto en otra sesión
usamos el comando load(’A’). Por ejemplo, con

sage: A = matrix([[1,2],[3,4]])
sage: A
[1 2]
[3 4]
sage: save(A,’A’)
sage: quit

Observatorio Astronómico de Quito-Escuela Politécnica Nacional 18



2.2 La Notebook Manual de cálculo científico en SageMath

se guarda la matriz A en un objeto llamado A. Si se cierra la sesión y se inicia una nueva, para cargar la
matriz A usamos el comando

sage: A = load(’A’)
sage: A
[1 2]
[3 4]

Para guardar sesiones usamos el comando save_session(’S’) donde S es el nombre con el cual
se guarda la sesión. Para cargar la sesión S guardada usamos el comando load_session(’S’). Por
ejemplo, con

sage: A = matrix([[1,2],[3,4]])
sage: b = 5
sage: save_session(’Ayb’)
sage: quit

se guarda la sesión con el nombre Ayb. Para cargar la sesión, es decir, en este caso la matriz A y el valor de
b, en otra sesión usamos

sage: load_session(’Ayb’)
sage: A
[1 2]
[3 4]

2.2. La Notebook
2.2.1. Crear una worksheet

Figura 3: La Notebook de SageMath

Para iniciar una sesión de Sage en la Notebook, desde la terminal interactiva lo hacemos con el co-
mando sage: notebook(). En Windows, cuando iniciamos la máquina virtual se abre por defecto la
Notebook. En la figura (3) se muestra la interfaz de la Notebook de SageMath.
El comando sage: notebook("directorio") inicia la Notebook y guarda los resultados de la se-
sión en el directorio especificado.

Para crear una worksheet, damos click en New Worksheet, ingresamos el nombre de la hoja de
trabajo, por ejemplo ’’Clase1’’, y damos click en Rename.
Los comandos se ingresan en las cajas de texto de la hoja de trabajo. Para evaluar las entradas lo podemos
hacer dando click en evaluate u oprimiendo Shift+Enter

Para guardar una worksheet damos click en Guardar o podemos guardarlo como archivo .sws al ir al
menú File... y luego click en Save worksheet to a file.... Tendremos que poner un título
a la hoja de trabajo y luego se descargará.
Dentro del menú Action... hay opciones que permiten borrar todas las salidas (Delete All Output...),
reiniciar la worksheet (Restart worksheet), evaluar todo (Evaluate All...), etc.
Dentro del menú File... adicional a guardar la worksheet como archivo .sws, también se la pue-
de imprimir (Print). Además es posible importar archivos y cargarlos a la worksheet con la opción
Load worksheet from a file... dentro del mismo meú.

2.2.2. Operaciones básicas

La mayoría de los comandos que se usaron en la terminal interactiva se los puede usar también en la
Notebook. Por ejemplo, para sumar dos números ingresamos 5+8 y evaluamos la entrada para mostrar el
resultado, así mismo con las demás operaciones.
Por ejemplo, si deseamos conocer el tiempo que el procesador estuvo activo ejecutándo alguna serie de
entradas, usamos igualmente que en la terminal interactiva, el comando cputime().

Si en la entrada halla algún error, nos mostrará el tipo de error que encontró y sugerirá su ubicación.
También se puede acceder a la ayuda de Sage con el comando nombre? donde nombre es el nombre del
comando del que queremos información, así también con el comando nombre?? se muestra su código
fuente. Por ejemplo, se puede ingresar diff? para obtener información del comando diff.

Cuando se termina con una línea de código se pasa a la siguiente línea al cambiar de caja de texto o al
oprimir Enter dentro de la misma caja.
Para hacer comentarios, es decir, líneas dentro de las cajas de texto que no deben ser evaluadas por Sage, se
debe comenzar la línea con el símbolo #.

2.2.3. Print y Show

Cuando en una caja de texto de la Notebook de Sage ingresamos más de una línea de entrada, después
de ejecutarla se mostrará en pantalla el resultado de la última operación, por ejemplo, si ingresamos

5+8
7*3

en pantalla se mostrará únicamente el resultado de evaluar 7*3 y no el de evaluar 5+8. Para que se muestre
en pantalla el resultado de las dos operaciones utilizamos el comando print(). Con el ejemplo anterior,
al ingresar

print(5+8)
print(7*3)

el resultado de la evaluación es

13
21

El comando show() realiza lo mismo que el comando anterior pero con la diferencia de que muestra
el resultado de una forma más agradable. Por ejemplo,

show(5+8)
show(7*3)

muestra
13
21

Este comando lo utilizaremos posteriormente para apreciar mejor la diferencia entre usar print() y
show().

2.3. Álgebra y cálculo elemental
A partir de esta sección se empleará la Notebook de Sage.

2.3.1. Definición de variables

SageMath maneja cálculo simbólico. Para esto es necesario declarar variables con el comando
var(’nombre’). Por ejemplo, con la siguiente línea asignamos a x la variable llamada x

x = var(’x’)

También se pueden hacer varias asignaciones en una misma entrada si separamos los objetos con comas,
por ejemplo con

a,b,c = var(’a,b,c’)

asigmanos a a, b y c las variables a, b y c, respectivamente.

2.3.2. Solución básica de ecuaciones

Para resolver ecuaciones en Sage se utiliza el comando solve(). Continuemos con un ejemplo. De-
claramos la variable x y creamos la función f(x) con las siguientes entradas

x = var(’x’)
f(x) = x^2 - 3*x -2

Para hallar las raíces de f(x), es decir, resolver la ecuación f(x) = 0, usamos el comando
solve(f(x), x), donde en el primer argumento se coloca la ecuación a resolver y en el segundo la
variable con respecto a la cual se resuelve la ecuación. Cuando la ecuación está igualada a cero, como en
este caso, no es necesario colocar la igualdad. Cabe recalcar que la comparación entre dos objetos en Sage
se la realiza con ==.

Siguiendo con el ejemplo, al evaluar la entrada solve(f(x), x), el resultado será

[x == -1/2*sqrt(17) + 3/2,
x == 1/2*sqrt(17) + 3/2]

Se puede usar el comando show() para la solución se muestre de forma más agradable, por ejemplo al
evaluar

show(solve(f(x), x))

se muestra como resultado [
x = −1

2

√
17 +

3

2
, x =

1

2

√
17 +

3

2

]
Si queremos presentar el resultado dentro de un documento hecho en LATEX, lo mejor es mostrar el resultado
con la sintaxis de LATEX. Para esto se usa el comando latex(). Por ejemplo, al evaluar

latex(solve(f(x), x))

el resultado es

\left[x = -\frac{1}{2} \, \sqrt{17} + \frac{3}{2},
x = \frac{1}{2} \, \sqrt{17} + \frac{3}{2}\right]

2.3.3. Substitución y simplificación

Para evaluar alguna función considerando un valor específico de alguna de sus variables se usa el co-
mando subs(). Para aclarar mejor su uso continuamos con el ejemplo anterior. Para evaluar la función
f(x) en x = −1/2 ∗

√
17 + 3/2 ingresamos la siguiente línea

r = f(x).subs(x=-1/2*sqrt(17) + 3/2)

El valor resultante de la evaluación se almacena en r. Para mostrarlo ingresamos show(r) y dado que
x = −1/2 ∗

√
17 + 3/2 es una raíz de f(x) esperaríamos que el resultado fuera 0, pero se obtiene

1

4

(√
17− 3

)2
+

3

2

√
17− 13

2
,

esto debido a que Sage realiza la substitución pero no simplifica. Para esto existen otros comandos que se
listan a continuación

expand()

canonicalize_radical()

simplify()

simplify_full()

Con el ejemplo, si se evalúa r.expand(), r.canonicalize_radical() o r.simplify_full()
el resultado es 0. Mientras que si se evalúa r.simplify() el resultado es

1

4

(√
17− 3

)2
+

3

2

√
17− 13

2

Los resultados no son siempre los mismos al evaluar los comandos de simplificación debido a los algo-
ritmos de los diferentes comandos.
Al evaluar el comando de simplificación el valor de la variable no cambia. Para que la variable cambie al
valor simplificado se debe realizar una asignación a otra variable.

2.3.4. Ecuaciones con constantes

Para trabajar con constantes, debemos declarar las mismas como variables. Por ejemplo, declaramos

a,b,c = var(’a,b,c’)
g(x) = a*x^2 + b*x + c

donde a, b, c se consideran constantes y la única variable de la función g(x) es x. Si mostramos la función
con show(g(x)) se tendrá

x 7→ ax2 + bx+ c

y si queremos hallar sus raices lo hacemos con el comando

show(solve(g(x),x)) [
x = −b+

√
b2 − 4 ac

2 a
, x = −b−

√
b2 − 4 ac

2 a

]
donde a, b, c son consideradas constantes.

2.3.5. Ecuaciones en varias variables

Así como se permite declarar constantes como variables, está permitido declarar más de una variable
para una misma función. Por ejemplo, si se ingresa

x,y = var(’x,y’)
eq = x*y + 3*y == 5*x + 2
show(eq)

xy + 3 y = 5x+ 2

se reconocerá como variable de eq a x y a y, por lo tanto es posible resolver la ecuación para cualquiera de
las dos variables, como se hace a continuación. Si se ingresa

show(solve(eq,y))

se mostrará [
y =

5x+ 2

x+ 3

]
que es la solución para y en función de x de la ecuación eq.

En cambio, si ingresamos

show(solve(eq,x))

se mostrará [
x = −3 y − 2

y − 5

]
que es la solución para x en función de y de la ecuación eq.
Para ecuaciones no lineales no es posible encontrar una solución para cada variable en todos los casos, por
lo que si no se encuentra una solución, en pantalla se mostrará la misma ecuación.

2.3.6. Sistema de ecuaciones

Los sistemas de ecuaciones se resuelven en Sage con el mismo comando solve(), pero en el primer
argumento se debe ingresar una lista con todas las ecuaciones y en los demás argumentos las variables con
respecto a las cuales se debe resolver el sistema. La solución será una lista de los valores de las variables
para los cuales el sistema de ecuaciones se resuelve. A continuación un ejemplo, ingresamos las ecuaciones
y las mostramos con el comando print():

eq1 = x + y == 6
print(eq1)
eq2 = x - 2*y == 3
print(eq2)

El resultado de la evaluación será

x + y == 6
x - 2*y == 3

Para resolver el sistema de dos ecuaciones (eq1 y eq2) con dos variables (x y y) usamos el comando

solve([eq1,eq2],x,y)

y se muestra

[[x == 5, y == 1]]

que es la única solución del sistema de ecuaciones.
Se recomienda, como se hizo en este ejemplo, asignar las ecuaciones previamente y no colocarlas di-

rectamente en el comando solve() dado que facilita la búsqueda de errores.
Las ecuaciones pueden ser manipuladas por medio de comandos, por ejemplo, si evaluamos eq1 y

luego ingresamos

_*2

estamos multiplicando la ecuación eq1, que es la última salida, por 2. La multiplicación se realiza a ambos
lados de la igualdad. También podemos ingresar

_+eq2

y la expresión resultante de la operación anterior se sumará a la ecuación eq2. Y si dividimos para tres la
expresión que resulta de la operación, es decir,

_/3

el resultado será

x == 5

que es el valor de x que resuelve el sistema. Para encontrar el valor de y se puede reemplazar x = 5 en la
ecuación eq1 y restarle 5 y obtener

eq1.subs(x=5)
_-5

y == 1

que es la solución del sistema. Sage también resuelve sistemas de ecuaciones no lineales. Por ejemplo,

p,q = var(’p,q’)
eq1 = p+q==9
eq2 = q*y+p*x==-6
eq3 = q*y^2+p*x^2==24
eq4 = q^2*x+p^2*y==5
sol = solve([eq1,eq2,eq3,eq4],p,q,x,y)
sol

[[p == (5.552416895784445 + 7.3729910979981*I), q == (3.447583104215555
- 7.3729910979981*I), x == (-1.388103961779249 + 1.199650705160947*I), y
== (0.151442392735851 + 1.36040251406414*I)], [p == (5.552416895784444 -
7.372991097998099*I), q == (3.447583104215554 + 7.3729910979981*I), x ==
(-1.388103961779249 - 1.199650705160947*I), y == (0.151442392735851 -
1.36040251406414*I)], [p == (3.447583104215556 - 7.3729910979981*I), q
== (5.552416895784443 + 7.372991097998102*I), x == (0.1514423927358509 +
1.360402514064139*I), y == (-1.388103961779249 + 1.199650705160947*I)],
[p == (3.447583104215556 + 7.372991097998099*I), q == (5.552416895784443
- 7.372991097998102*I), x == (0.151442392735851 - 1.360402514064139*I),
y == (-1.388103961779249 - 1.199650705160947*I)], [p ==
5.294879340788699, q == 3.705120659211301, x == -1.913667953667954, y ==
1.115386174741537], [p == 3.705120659211301, q == 5.294879340788699, x
== 1.115386174741537, y == -1.913667953667954]]

donde se tiene una lista de soluciones al sistema de ecuaciones no lineales.
Para que no se muestre toda la lista de soluciones del sistema, se puede manipular la salida. En el

ejemplo, en sol se guardan todas las soluciones al sistema del ecuaciones planteado, pero si queremos que
se muestre únicamente la primera solución deberemos ingresar

sol[0]

[p == (5.552416895784445 + 7.3729910979981*I),
q == (3.447583104215555 - 7.3729910979981*I),
x == (-1.388103961779249 + 1.199650705160947*I),
y == (0.151442392735851 + 1.36040251406414*I)]

dado que la lista de soluciones se numeran desde el cero. Y se puede realizar lo mismo con las demás
soluciones, por ejemplo con la tercera solución

sol[2]

[p == (3.447583104215556 - 7.3729910979981*I),
q == (5.552416895784443 + 7.372991097998102*I),
x == (0.1514423927358509 + 1.360402514064139*I),
y == (-1.388103961779249 + 1.199650705160947*I)]

Dentro de cada solución hay una lista de valores, en este caso los valores de p, q, x, y que resuelven el
sistema. Si queremos que se muestre únicamente el valor de p de la primera solución deberemos ingresar

sol[0][0]

p == (5.552416895784445 + 7.3729910979981*I)

y así con todos los elementos de la lista. Cabe recordar que los elementos de las listas están numerados
desde el cero.

También es posible mostrar sólo un lado de alguna ecuación con los comandos lhs() para mostrar
el lado izquierdo y rhs() para mostrar el lado derecho. Por ejemplo, en la última salida del ejemplo, si
queremos sólo conocer el valor de p de la primera solución, ingresamos

sol[0][0].rhs()

5.552416895784445 + 7.3729910979981*I

2.3.7. Soluciones numéricas

En ocasiones no se pueden encontrar soluciones analíticas para una ecuación por lo que es necesario re-
currir a métodos numéricos que mostrarán soluciones aproximadas dentro de un margen de error. SageMath
permite resolver ecuaciones de forma numérica con el comando find_root(). Se debe recordar que para
resolver de forma numérica no se deben tener constantes indeterminadas. Todas las constantes deben tener
una valor establecido. A continuación, con un ejemplo, detallamos como usar el comando find_root().

Si buscamos resolver la ecuación cos(x) = sin(x) con el comando solve() en pantalla se mostrará
[sin(x) == cos(x)]

que corresponde a la misma ecuación dado que Sage no encuentra una solución analítica. Para resolver la
ecuación de forma numérica evaluamos

find_root(cos(x)==sin(x),0,pi/2)
donde el primer argumento es la ecuación, los siguientes argumentos son el intervalo en el cual se va a
buscar la solución: el segundo argumento es el valor de x desde donde se va a buscar una solución y el
tercer argumento el valor x hasta el cual se va a realizar la búsqueda. En este caso, si se evalúa

find_root(cos(x)==sin(x),0,pi/2)
se obtendrá 0.7853981633974484 como solución de la ecuación. Para comprobar el error de la solu-
ción se puede evaluar

cos(find_root(cos(x)==sin(x),0,pi/2))-sin(find_root(cos(x)==sin(x),0,pi/2))

que resulta
5.55111512312578e-17

que es el error del método numérico dado que si fuera una solución exacta, el resultado de la evaluación
debería ser cero.

2.3.8. Diferenciación

Sage realiza diferenciación simbólica por medio del comando diff(). Para esto, dentro de los argu-
mentos se debe colocar la función a diferenciar, las variables con respecto a la cual se realiza la diferencia-
ción y el orden de la diferenciación. Por ejemplo, para determinar la primera derivada de sin(x) se evalúa
el comando

diff(sin(x),x)
y muestra

cos(x)
Para derivadas de orden superior, por ejemplo, la tercera derivada de sin(x^2), se evalúa el comando

diff(sin(x^2),x,x,x)
que resulta

-8*x^3*cos(x^2) - 12*x*sin(x^2)

También se puede evaluar el comando
diff(sin(x^2),x,3)

donde el tercer argumento es el orden de la derivación.
Si se tiene una función en varias variables la derivación se la realiza especificando la variable con

respecto a la cual se deriva, por ejemplo

f(x,y) = x^2 + 17*y^3
f.diff(x)

determina la primera derivada de f(x, y) respecto a x y el comando

f(x,y).diff(y,2)

la segunda derivada de f(x, y) respecto a y.

2.3.9. Integración

Sage realiza integración simbólica por medio del comando integral(). Entre los argumentos del
comando se debe colocar la función a integrar y la variable de integración. Si se coloca también argumentos
como los límites de integración, realiza integración definida. Si Sage no encuentra una primitiva, muestra
la misma integral. Por ejemplo,

show(integral(x*sin(x^2),x))

−1

2
cos
(
x2
)

integra x sin(x2) con respecto a x y

show(integral(x/(x^2+1),x,0,1))

1

2
log (2)

integra
x

x2 + 1
con respecto a x entre 0 y 1.

Dado que no siempre puede hallar una primitiva, para las integrales definidas también cuenta con un
comando para integración numérica. El comando es numerical_integral() donde se colocan como
argumentos la función y los límites de integración. El resultado de la evaluación son dos números: el primero
es el resultado de la integración y el segundo es una cota al error numérico cometido. Por ejemplo, al realizar
la misma integral definida

x

x2 + 1
con respecto a x entre 0 y 1 al evaluar el comando

numerical_integral(x/(x^2+1),0,1)

se obtiene (0.3465735902799726, 3.847739796558309e-15). Con la integración no numéri-
ca se obtuvo 1

2
log (2) ≈ 0,34657359027997264.

2.3.10. Ecuaciones diferenciales

Para resolver ecuaciones diferenciales se debe usar el comando desolve() donde como argumen-
tos se coloca la ecuación diferencial a resolver y la función que se quiere hallar. Se debe especificar la
dependencia de la función. Por ejemplo, con el comando

t = var(’t’)

se define t como variable y luego con

x = function(’x’,t)

se define la dependencia de x respecto al tiempo. Definimos una ecuación diferencial

eqDif = diff(x,t) + x - 1

y la resolvemos con el comando

show(desolve(eqDif,x)) (
C + et

)
e(−t)

que es la función que resuelve la ecuación diferencial y C es una constante de integración. Se pueden dar
condiciones a la ecuación diferencial aumentando un argumento en el comando desolve() y con esto se
deja sin arbitrariedad las constantes de integración. Lo ilustramos con el ejemplo anterior. Si se quiere que
la solución además cumpla con que al ser evaluada en t = 10 su valor sea 2 se debe evaluar

show(desolve(eqDif,x,ics=[10,2]))(
e10 + et

)
e(−t)

que es la solución a la ecuación diferencial que cumple con la condición impuesta.
También se puede hallar una solución por métodos numéricos con el comando desolve_odeint()

que resuelve ecuaciones diferenciales ordinarias de primer orden. Se ilustra el uso del comando con un
ejemplo a continuación

x = var(’x’)
epsilon = 0.00001
f = -x+1
ic = 2
t = srange(10,11+epsilon,epsilon)
sol=desolve_odeint(f,ic,t,x)
sol

En la primera línea se declara la función que se quiere hallar, en la segunda el paso de la integración
numérica, en la tercera establece la ecuación diferencial escrita de tal forma que la derivada de primer
orden esté en el lado izquierdo y el resto al lado derecho de la igualdad, en la cuarta línea se estable el
valor de la función en el punto inicial y en la quinta línea se crea un arreglo que es un intervalo desde
el punto inicial hasta el final en donde se va a evaluar la función solución. En la sexta línea se resuelve la
ecuación diferencial: en el primer argumento se coloca la ecuación diferencial, en el segundo las condiciones
iniciales, en el tercero el intervalo y en el cuarto la función a determinar. El resultado sol es

array([[ 2. ],
[ 1.99999 ],
[ 1.99998 ],
...,
[ 1.36788683],
[ 1.36788315],
[ 1.36787947]])

un arreglo donde están los valores de x para los distintos valores de t en el intervalo.

2.3.11. Límites

Los límites se pueden obtener con el comando limit() y en los argumentos son principalmente la

función y el valor al cual tiende la variable independiente. Por ejemplo, para obtener el límite de
sin(x)

x
cuando x→ 0 usamos el comando

lim(sin(x)/x,x=0)

el resultado será 1. También se pueden calcular límites al infinito por ejemplo

lim(sin(x)/x,x=oo)

cuyo resultado es 0.
Adicionalmente se puede incluir un argumento donde se indique si el límite se lo toma por la izquierda

o por la derecha, como se ilustra con el siguiente ejemplo

lim(1/x,x=0,dir=’minus’)

donde se toma el límite de 1/x cuando x → 0 por la izquierda y el resultado es −∞ y también se puede
tomar el límite por la derecha con

lim(1/x,x=0,dir=’plus’)

cuyo valor es +∞.

2.3.12. Series de Taylor

Se pueden hacer expansiones en series de Taylor por medio del comando taylor() donde se debe
colocar la función a expandir, la variable con respecto a la cual se expande, el punto alrededor del cual se
hace la expansión y el orden. Por ejemplo, si se quiere expandir sin(x) alrededor de x = 0 hasta un quinto
orden, usamos el comando

show(taylor(sin(x),x,0,5))

1

120
x5 − 1

6
x3 + x

Si el desarrollo se lo hace alrededor de x = 1 y hasta un tercer orden, usamos

show(taylor(sin(x),x,1,3))

−1

6
(x− 1)3 cos (1)− 1

2
(x− 1)2 sin (1) + (x− 1) cos (1) + sin (1)

2.3.13. Suma de series

Con el comando sum() se puede determinar la suma de una serie. Lo ilustramos con ejemplos. Si se
quiere determinar el valor de

5∑
n=1

1

2n

usamos el comando

n = var(’n’)
sum(1/2^n,n,1,5)

y el resultado es 31/32.
En Sage también se pueden determinar el valor de la suma de una serie infinita. Por ejemplo, para

determinar el valor de
∞∑
n=1

1

2n

evaluamos

sum(1/2^n,n,1,oo)

y el resultado es 1.
No todas las sumas al infinito son convergentes. Por ejemplo,

∞∑
n=1

1

n

es divergente. En Sage, si la suma es divergente, se muestra un mensaje que indica este resultado. Por
ejemplo, con el comando

sum(1/n,n,1,oo)

muestra

#0: simplify_sum(expr=’sum(1/_SAGE_VAR_n,_SAGE_VAR_n,1,inf))
Traceback (click to the left of this block for traceback)
...
ValueError: Sum is divergent.

Indicamos otros ejemplos a continuación. La suma
∞∑
n=1

1

n2

se puede evaluar con

sum(1/n^2,n,1,oo)

y el resultado será 1
6
π2.

Por último, si queremos evaluar una sumatoria aparentemente más complicada como
∞∑
n=0

1

(2n+ 3)2(2n+ 1)2(2n− 1)2

en Sage, también la podemos evaluar con

sum(1/((2*n-1)^2*(2*n+1)^2*(2*n+3)^2), n, 0, oo)

y el resultado será 3
256

π2.

2.4. Gráficos en 2D
2.4.1. Objetos gráficos

Sage tiene una serie de comandos que permiten graficar objetos como:

Puntos: Para graficar puntos en un plano cartesiano se lo puede hacer por medio del comando
point(). Se debe incluir en los argumentos la lista de puntos que se quieren graficar. La lista
debe estar entre corchetes y los puntos entre paréntesis, separando las coordenadas por medio de una
coma. Por ejemplo, con el comando point([(1,1)]) se grafica el punto (1,1). En la figura (4)
se muestra el gráfico del punto (1, 1) que resulta de la evaluación del comando anterior. Si queremos
además graficar el punto (1,2) deberemos usar el comando point([(1,1),(1,2)]).

0.5 1 1.5 2

0.5

1

1.5

2

Figura 4: Punto (1,1)

Círculos: Para graficar un círculo se puede usar el comando circle(). Como primer argumento se
debe indicar el punto centro del círculo y como segundo argumento el radio del mismo. Por ejemplo,
para graficar un círculo centrado en el origen y de radio 1, evaluamos el comando circle((0,0), 1).
El gráfico resultante se muestra en la figura (5).

Figura 5: Círculo centrado en el origen de r = 1

Se pueden incluir argumentos adicionales como color=’’ para cambiar el color predeterminado de la
imagen y fill=’’ para pintarla. El color por defecto es el azul. Se recomienda leer la ayuda del comando
(por ejemplo, circle?) para más información sobre las características del gráfico que pueden cambiarse.
Por ejemplo, con el comando
circle((0.5,0.5), 0.5, color=’red’, fill=True)
graficamos un círculo centrado en (0.5,0.5) de radio 0.5 pintado de color rojo. El gráfico resultante se
muestra en la figura (6).

Figura 6: Círculo centrado en (0,5, 0,5) de r = 0,5

Sage permite incluir en un mismo gráfico varios objetos. Esto se hace mediante una suma de gráficos.
Por ejemplo, si queremos que todos los gráficos que hicimos anteriormente estén en un mismo plano,
primero asignamos los objetos de la siguiente forma

p1 = point([(1,1)])
p2 = circle((0,0), 1)
p3 = circle((0.5,0.5), 0.5, color=’red’, fill=True)

Después, mostramos la suma de los tres gráficos con el comando show(p1+p2+p3) y el resultado será
un gráfico con los tres objetos. Ver la figura (7).

Figura 7: Suma de objetos gráficos

2.4.2. Listas

Con el comando list_plot() se puede graficar un conjunto de puntos. Por ejemplo, con el comando
list_plot([(0, 3.5), (1, 2), (2, 5.1)])
se grafican todos los puntos que están dentro de la lista. Si queremos que los puntos se unan uno a uno con
una recta se agrega el argumento plotjoined=True, es decir,

list_plot([(0, 3.5), (1, 2), (2, 5.1)],plotjoined=True)

El gráfico resultante se muestra en la figura (8).
En lugar de escribir la lista como puntos coordinados, se puede usar zip() para escribir por un lado

los valores del eje x y por otro lado los valores del eje y. Por ejemplo, con el ejemplo anterior, el comando

list_plot(zip([0,1,2], [3.5,2,5.1]))

da el mismo gráfico que el que resulta con el comando
list_plot([(0, 3.5), (1, 2), (2, 5.1)]).

Observatorio Astronómico de Quito-Escuela Politécnica Nacional 31



2.4 Gráficos en 2D Manual de cálculo científico en SageMath

Figura 8: Lista de puntos unidos con rectas

2.4.3. Funciones

Con el comando plot() se pueden graficar funciones. En los argumentos se debe incluir la función y
el rango en el cual se debe graficar. Por ejemplo, para graficar la función cos(x) entre −5 y 5, evaluamos la
entrada plot(cos, -5, 5). En este caso no es necesario recalcar la variable independiente de la fun-
ción. Si se incluye la variable en la función, es necesario indicar la variable en el comando plot(). Para el
mismo ejemplo, el comando plot(cos(x),x,-5,5) da el mismo gráfico que plot(cos, -5, 5).
Ver el gráfico resultante en la figura (9).

Figura 9: Función Coseno

En lugar de incluir el intervalo en el cual se debe graficar la función se pueden establecer los valores
máximos y mímos para x y y. Esto se hace al incluir en el comando plot() los argumentos xmin=’’,
xmax=’’, ymin=’’ y ymax=’’. Por ejemplo, se puede graficar

plot(cos(x),x,xmin=-4,xmax=6,ymin=0,ymax=1)
y se obtiene el gráfico de la figura (10).

2.4.4. Gráficos Paramétricos

Para hacer gráficos parametrizados se usa el comando parametric_plot(). En los argumentos se
deben colocar, en forma de par ordenado, los valores de x y y parametrizados, y adicionalmente el parámetro
y el intervalo del mismo. Para ilustrar el uso de este comando, graficaremos un círculo de radio 1 parame-
trizado. Para esto, se sabe que, si t es el parámetro, la parametrización x(t) = cos(t) y y(t) = sin(t) cuando
0 ≤ t ≤ 2π, resulta un círculo de radio 1. Entonces, para graficarlo primero declaramos como variable al pa-
rámetro con t=var(’t’) y evaluamos parametric_plot((cos(t),sin(t)),(t,0,2*pi)).
Ver la figura (11).

Figura 10: Función Coseno con ejes cortados

Figura 11: Gráfico de función parametrizada

2.4.5. Gráficos polares

Para gráficos en coordenadas polares se usa el comando polar_plot(). En los argumentos de debe
colocar la expresión de r(θ), es decir, del radio en función de ángulo, y adicional se debe incluir el intervalo
de los valores del ángulo. Por ejemplo, para graficar r(θ) = sin2(3θ) cuando 0 ≤ θ ≤ 2π, declaramos la
variable x (x=var(’x’)) y el comando polar_plot(sin(3*x)^2, (x, 0, 2*pi)). El gráfico
se muestra en la figura (12).

Figura 12: Gráfico en polares

2.4.6. Gráficos implícitos

Mediante el comando implicit_plot() se pueden graficar curvas de nivel. Para esto se debe
ingresar la función de dos variables f(x, y) e igualarla a una constante. Por defecto grafica la curva

Observatorio Astronómico de Quito-Escuela Politécnica Nacional 33



2.4 Gráficos en 2D Manual de cálculo científico en SageMath

f(x, y) = 0. También se debe ingresar los rangos de las variables x y y. Por ejemplo, si se quiere
graficar las curvas de nivel de f(x, y) = x2 − y2, en especial cuando f(x, y) = 1, en el rango
−3 ≤ x, y ≤ 3, se evalúa el comando

implicit_plot(x^2-y^2==1, (x, -3, 3), (y, -3,3))
Ver la figura (13).

Figura 13: Curva de nivel x2 − y2 = 1

Es posible hacer gráficos de las líneas de contorno, es decir, de las líneas donde el valor de f(x, y) es
constante, con el comando contour_plot(). Para esto, siguiendo con el mismo ejemplo, evalua-
mos

contour_plot(x^2-y^2, (x,-3, 3), (y,-3,3))
El resultado es el gráfico de la figura (14) en el cual, el color blanco representa los lugares donde la
función f(x, y) es mayor.

Figura 14: Líneas de contorno de x2 − y2

Otra forma de presentar las curvas de nivel es por medio de gráficos de densidad. En Sage se los
realiza con el comando density_plot(). Continuando con el mismo ejemplo, evaluamos

density_plot(x^2-y^2,(x,-2,2),(y,-2,2),cmap=’coolwarm’)
para obtener un gráfico de densidad de f(x, y) = x2− y2. El argumento cmap es la paleta de colores
que se usará en el gráfico. Ver el gráfico de la figura (15). En el gráfico, el color rojo representa los
lugares donde la función f(x, y) es mayor.

2.4.7. Campos vectoriales

Se pueden graficar campos vectoriales con el comando plot_vector_field(). Este comando to-
ma dos funciones de dos diferentes variables y grafica arreglos vectoriales de la función en los rangos

Figura 15: Gráfico de densidad de x2 − y2

que se establezcan para las variables. Por ejemplo, se puede graficar el gradiente del ejemplo anterior
(f(x, y) = x2 − y2). Para esto sabemos que∇f(x, y) = (2x,−2y), por lo tanto con el comando

plot_vector_field((2*x,-2*y),(x,-3,3),(y,-3,3))
graficamos el campo vectorial perpendicular a las curvas de nivel de f(x, y) = x2 − y2. El resultado se
muestra en la figura (16).

Figura 16: Gradiente de x2 − y2

Se puede hacer un gráfico para comprobar que el gradiente es perpendicular a las curvas de nivel. Para
esto juntamos en un mismo gráfico dos de los gráficos hechos en los ejemplos anteriores. Por ejemplo,
evaluemos

con = contour_plot(x^2-y^2, (x,-3, 3), (y,-3, 3))
gra = plot_vector_field((2*x,-2*y),(x,-3,3),(y,-3,3))
con + gra

y obtendremos el gráfico de la figura (17).

2.4.8. Manipulación de gráficos

Se cuenta con una serie de comandos que permiten manipular los gráficos en Sage. Se enlistan algunos
comandos:

aspect_ratio(): se utiliza para cambiar la relación entre las escalas en los ejes. Se pone como
ejemplo las siguientes entradas

gra = plot(cos(x),(x,-5,5))
gra.set_aspect_ratio(0.5)
show(gra)

Figura 17: Gradiente perpendicular a las curvas de nivel

donde la escala del eje y se reduce a la mitad comparada con la escala en el eje x.

axes_labels(): se utiliza para ponerle nombre a los ejes. Se puede usar la sintaxis de LATEX para
los nombres. Por ejemplo,

gra.axes_labels([’$x$’,’$y$’])

le cambia los nombres a los ejes del ejemplo anterior. El gráfico manipulado se muestra en la figura
(18).

Figura 18: Gráfico cos(x)

save(): se utiliza para guardar un gráfico. Se debe colocar como argumento el directorio donde se
guardará el archivo y el nombre del mismo con la extensión (.png, .eps). Por ejemplo, para guardar el
gráfico del ejemplo anterior usamos el comando

gra.save(’/home/oaq-astronomia1/Escritorio/coseno.eps’)
Con esto, el gráfico se guardará en el Escritorio con el nombre coseno y será un archivo .eps.

2.5. Gráficos en 3D
Para graficar en 3D se tienen algunos comandos.

plot3d(): Se utiliza para graficar funciones con dos variables. En sus argumentos se debe estable-
cer la función y los rangos para las variables. Por ejemplo, para graficar f(x, y) = x2 − y2 cuando
−2 ≤ x, y ≤ 2, evaluamos

x,y=var(’x,y’)
f(x,y)=x^2-y^2
plot3d(f(x,y),(x,-2,2),(y,-2,2))

El gráfico se muestra en la figura (19).

Figura 19: f(x, y) = x2 − y2

implicit_plot3d(): Se utiliza para graficación implícita. Al igual que en dos dimensiones, se
lo utiliza para graficar las curvas de nivel. Se necesita una función de tres variables igualada a una
constantes. Por ejemplo, para obtener la curva de nivel igual a cero de f(x, y, z) = x2 + y2 − z2

evaluamos

x,y,z=var(’x,y,z’)
implicit_plot3d(x^2+y^2-z^2==0,(x,-3,3),(y,-3,3),(z,-3,3))

El resultado se muestra en la figura (20).

Figura 20: Curva de nivel f(x, y, z, w) = x2 − y2 − z2 = 0

parametric_plot3d(): Se utiliza para gráficos paramétricos. Se deben colocar las parametri-
zaciones de las variables x, y, z y el rango del parámetro. Por ejemplo, para graficar una espiral, su
parametrización en función de z es x(z) = cos(z), y(z) = sin(z) y z(z) = z. Por lo tanto usamos el
comando

parametric_plot3d((cos(z),sin(z),z),(z,-5,5))

El resultado se muestra en la figura (21).

plot_vector_field3d(): Se usa para graficar campos vectoriales en tres dimansiones. Se debe
detallar el campo vectorial en forma de triada ordenada y los rangos de las variables. Por ejemplo,

Figura 21: Espiral parametrizado

Figura 22: Campo vectorial (y,−x, z)

plot_vector_field3d((y,-x,z),(x,-1,1),(y,-1,1),(z,-1,1))

El resultado se muestra en la figura (22).

Para grabar las gráficas no se puede usar el comando save(). Para grabar las gráficas se debe dar click
Derecho sobre la imagen generada y en el menú buscar Exportar y dar click en el formato. Dentro de
este menú hay opciones para manipular las gráficas.

2.6. Vectores y Matrices
2.6.1. Operaciones con vectores

Declaración: Con el comando vector() se declaran vectores. Como argumento se deben colocar
sus componentes en forma de lista. Por ejemplo, con el comando v = vector([1,1,-4]) se
declara el vector v de dimensión 3 y componentes (1,1,-4).

Suma: Para sumar vectores, componente a componente, se usa el operador +. Si la dimensión de los
vectores que se suman no es la misma, el programa muestra un error. A continuación dos ejemplos.
Para sumar los vectores v=(1,1,-4) y w=(3,0,2), usamos los comandos

v = vector([1,1,-4])
w = vector([3,0,2])
show(v + w)

y el resultado será
(4, 1, −2)

Por otro lado, si evaluamos

x= vector([1,1,-4])
y= vector([3,0,2,6])
show(x+y)

se mostrará

Traceback (click to the left of this block for traceback)
...
TypeError: unsupported operand parent(s) for ’+’: ’Ambient free module
of rank 3 over the principal ideal domain Integer Ring’ and ’Ambient
free module of rank 4 over the principal ideal domain Integer Ring’

Escalar por vector: Para multiplicar un escalar por un vector se usa el operador * y el resultado es
un vector cuyas componentes son producto de multiplicar el escalar por cada componente del vector
original. Por ejemplo, para multiplicar el vector v por 3 usamos el comando show(3*v).

Se pueden hacer combinaciones lineales. Por ejemplo, al evaluar el comando show(3*v - 2*w)
en pantalla se mostrará

(−3, 3, −16)

que es el vector resultante de la combinación lineal entre los vectores v y w.

Producto interno: Para el producto interno euclídeo entre dos vectores se usa el mismo operador *.
El resultado es un escalar. Por ejemplo, para realizar el producto interno entre los vectores v y w,
usamos el comando show(v*w) y el resultado será −5

2.6.2. Operaciones con matrices

Declaración: Para declarar una matriz en Sage usamos el comando Matrix(). Por ejemplo, al
evaluar el comando

A = Matrix([[1,2,3],[3,2,1],[1,1,1]])
estamos creando la matriz

A =

 1 2 3
3 2 1
1 1 1


es decir, el ingreso de los elementos de las matrices se realiza por filas y en forma de listas. Declare-
mos la matriz

B =

 1 2 3
3 2 1
2 2 5


con el comando B = Matrix([[1,2,3],[3,2,1],[2,2,5]]).

Suma: Para sumar dos matrices usamos el operador + y la suma se realizará sumando las componentes
de las matrices. Para poder sumar dos matrices, sus dimensiones deben ser las mismas. Por ejemplo,
sumemos las matrices A y B. Esto lo hacemos evaluando en comando A+B y se mostrará la matriz
resultante de la operación

[2 4 6]
[6 4 2]
[3 3 6]

Escalar por matriz: Es posible multiplicar un escalar por una matriz. Esto se lo realiza con el ope-
rador *. De esta forma es posible realizar combinaciones lineales de matrices. Por ejemplo, con el
comando 3*A - 2*B se mostrará

[ 1 2 3]
[ 3 2 1]
[-1 -1 -7]

que es la matriz resultante de la combinación lineal.

Producto de matrices: Con el mismo operador * es posible multiplicar dos matrices. Las dimensio-
nes de las matrices deben ser compatibles para poder realizar la operación. Por ejemplo, multiplique-
mos las matrices A y B. Este se lo hace con el comando A*B y el resultado será

[13 12 20]
[11 12 16]
[ 6 6 9]

Se puede comprobar que la multiplicación de matrices no es conmutativa. Evaluando el comando
A*B-B*A el resultado es

[ 3 3 12]
[ 1 1 4]
[-7 -7 -4]

que es distinto de la matriz nula, con lo cual, la operación no es conmutativa.

Matriz transpuesta: Con el comando transpose() se encuentra la matriz transpuesta. Por ejem-
plo, para hallar la matriz transpuesta de A evaluamos el comando A.transpose() y el resultado
será

[1 3 1]
[2 2 1]
[3 1 1]

Kernel de una matriz: Con el comando kernel() se encuentra el núcleo por la izquierda de la
matriz. En los argumentos se debe colocar la matriz. Por ejemplo, con el comando kernel(A)
estamos encontrando los vectores x que cumplen con

xA = 0

Si evaluamos el comando kernel(A) obtendremos

Free module of degree 3 and rank 1 over Integer Ring
Echelon basis matrix:
[ 1 1 -4]

es decir, si se multiplica por la izquierda cualquier combinación lineal del vector (1, 1,−4) con la
matriz A, el resultado será cero. Para comprobar este resultado evaluemos vector([1,1,-4])*A
y el resultado será

(0, 0, 0)

Para determinar el núcleo por la derecha se debe hallar el núcleo de la matriz transpuesta. Es decir,
para hallar los vectores x que cumplen con

Ax = 0

evaluamos el comando kernel(A.transpose()) y el resultado es

Free module of degree 3 and rank 1 over Integer Ring
Echelon basis matrix:
[ 1 -2 1]

y se lo puede comprobar el evaluar A*vector([2,-4,2]) cuyo resultado es

(0, 0, 0)

donde el vector (2,−4, 2) es una combinación lineal de los elementos de la base del núcleo, es decir,
una combinación lineal del vector (1,−2, 1).

Sistema de ecuaciones: Para resolver un sistema de ecuaciones del tipo

Ax = y

donde A es una matriz conocida, y un vector conocido y x es un vector cuyas componentes son las
incógnitas, se utiliza el comando solve_right(). Procedemos con un ejemplo. Para resolver el
sistema de tres ecuaciones con tres incógnitas 1 2 3

3 2 1
1 1 1

 a
b
c

 =

 0
−4
−1


evaluamos las entradas

y = vector([0, -4, -1])
x = A.solve_right(y)
x

y el resultado será

(-2, 1, 0)

es decir, la solución al sistema de ecuaciones es a = −2, b = 1 y c = 0.

Por otro lado, si queremos resolver un sistema de ecuaciones del tipo

xA = y

donde las componentes de x son las incógnitas, usamos el comando solve_left().

Determinante: Para calcular el determinante de una matriz usamos el comando determinant().
Por ejemplo, para calcular el determinante de la matriz B evaluamos la entrada B.determinant()
y el resultado será −12

Inversa: Para hallar la matriz inversa de una matriz usamos el comando inverse(). Por ejemplo,
para determinar la inversa de la matriz B evaluamos B.inverse() y se mostrará

[ -2/3 1/3 1/3]
[13/12 1/12 -2/3]
[ -1/6 -1/6 1/3]

También es posible determinar la inversa de una matriz usando ^(-1), es decir, la inversa de la
matriz B puede determinarse al evaluar B^(-1) y el resultado será el mismo.

Si la matriz es singular, Sage mostrará un mensaje de error en la pantalla al intentar hallar su inversa.
Por ejemplo, se puede comprobar que el determinante de la matriz A es igual a cero y por lo tanto es
una matriz singular. Si se evalúa A.inverse() se mostrará el error

Traceback (click to the left of this block for traceback)
...
ZeroDivisionError: Matrix is singular

Vectores y valores propios: Para determinar los valores propios de una matriz se usa el coman-
do eigenvalues(). Por ejemplo, los valores propios de la matriz A se los determina evaluando
A.eigenvalues(). El resultado es

[5, 0, -1]

que es una lista de los valores propios de la matriz. Para los vectores propios se debe establecer si se
desea establecer los vectores propios por la izquierda o por la derecha. Los comandos a utilizar son
eigenvectors_left() y eigenvectors_right() respectivamente.

Para la matriz del ejemplo, los vectores propios por la izquierda se determinan con el comando
show(A.eigenvectors_left())
y el resultado será[

(5, [(1, 1, 1)] , 1) , (0, [(1, 1, −4)] , 1) ,

(
−1,

[(
1, −1

5
, −7

5

)]
, 1

)]
En el resultado se muestra una lista de elementos. Dentro de cada elemento se encuentra como primer
valor el valor propio, luego el vector propio correspondiente a dicho valor propio y por último la
multiplicidad.

Matriz escalonada: La matriz escalonada de una determinada matriz se la obtiene al evaluar el co-
mando echelon_form(). Por ejemplo, al evaluar A.echelon_form() se obtendrá

[ 1 0 -1]
[ 0 1 2]
[ 0 0 0]

que es la matriz escalonada que se obtiene de A.

Matrices simbólicas: Hasta ahora hemos utilizado matrices cuyos componentes son números, pero
en Sage también es posible operar con matrices cuyos elementos sean funciones o variables. Por
ejemplo, podemos establecer el polinomio característico de la matriz A con las siguientes entradas

t = var(’t’)
id = matrix.identity(3)
pol = (A-t*id).determinant()
show(pol.expand())

Se obtendrá el polinomio −t3 + 4 t2 + 5 t. En la primera línea se estable la variable del polinomio.
En la segunda se crea el objeto id que es la matriz identidad de dimensión 3. En la tercera línea se
encuentra el polinomio característico pol(t) de la matriz A cuya definición es

pol(t) = det(A− t ∗ id)

La matriz t ∗ id es una matriz cuyos componentes son variables y dependen de t.

Si se desean hallar los valores propios se deben hallar las raices del polinomio característico. Para
esto evaluamos el comando solve(pol,t) cuyo resultado es

[t == 5, t == -1, t == 0]

que son los valores propios de la matriz A.

2.7. Espacios vectoriales
Para declarar espacios vectoriales se usa el comando VectorSpace(). En los argumentos se de-

be colocar el campo sobre el cual se define el espacio vectorial y la dimensión del mismo. Por ejem-
plo, para declarar un espacio vectorial tridimensional sobre los números racionales usamos el comando
V = VectorSpace(QQ,3). Otros campos que se pueden usar son los números complejos (CC) y los
números reales (RR).

Para mostrar la base del espacio vectorial V usamos el comando basis(). Para este caso, evaluando
show(V.basis()) se muestra

[(1, 0, 0) , (0, 1, 0) , (0, 0, 1)]

que es la base cartesiana del espacio vectorial. Los elementos de la base estan numerados a partir del cero.
Por ejemplo, si queremos acceder al primer elemento de la base unamos V.0, y así mismo con los demás
elementos.

Se pueden generar subespacios vectoriales con el comando subspace(), en el cual se deben especifi-
car los vectores que generan el subespacio. Por ejemplo, se puede establecer el subespacioW de V generado
por los vectores (1, 2, 7) y (1, 1, 0) con el comando W = V.subspace([[1,2,7], [1,1,0]]). Al
evaluar esta entrada se mostrará

Vector space of degree 3 and dimension 2 over Rational Field
Basis matrix:
[ 1 0 -7]
[ 0 1 7]

es decir, Sage establece otra base que genera el mismo subespacio W . Si queremos que los vectores que
ingresamos como generadores del subespacio se mantengan como elementos de la base usamos el comando
subspace_with_basis().

Se pueden realizar combinaciones lineales con los elementos de la base con el comando
linear_combination_of_basis(). Por ejemplo, si queremos hacer una combinación lineal con
los elementos de la base del subespacio vectorial W evaluamos
W.linear_combination_of_basis([1,1]). En sus argumentos se debe colocar en forma de lista
los coeficientes de la combinación lineal. El primer elemento de la lista se multiplica por el primer elemento
de la base y así sucesivamente. En este caso, el resultado de la combinación lineal es el vector (1, 1, 0).

Para declarar vectores dentro del espacio vectorial ya creado V , usamos V(). Por ejemplo, creemos los
vectores v y w con las siguientes entradas

v = V([1,1,-4])
w = V([3,0,2])

donde se deben especificar las componentes de los vectores. Para realizar el producto interno entre los dos
vectores usamos inner_product(). Por ejemplo, para hallar el producto interno entre v yw, evaluamos
v.inner_product(w) y el resultado es −5.

Si deseamos un elemento aleatorio del espacio vectorial usamos el comando random_element().
Por ejemplo, si queremos un elemento aleatorio del espacio vectorial V evaluamos V.random_element()
y se mostrará

(-1, -4, -1)

que es un vector que pertenece a V .
Si deseamos saber si dos vectores son linealmente dependientes o no en algún espacio vectorial, pode-

mos usar el comando are_linearly_dependent(). Por ejemplo, si queremos saber si los vectores v
y w son linealmente dependientes en V evaluamos V.are_linearly_dependent([v,w]) y en este
caso se mostrará

False

dado que son linealmente independientes.
Otros comando útiles para espacios vectoriales son dimension() que muestra la dimensión del espa-

cio vectorial, zero() que muestra el elemento cero del espacio vectorial y complement() que muestra
el subespacio vectorial complemento del algún subespacio.

2.8. Combinatoria
Para establecer el número de subconjuntos con k elementos que se pueden armar a partir de un conjunto

de n elementos se debe determinar el coeficiente binomial que está definido por(
n
k

)
=

n!

k!(n− k)!

En Sage se puede hacer este cálculo con la función binomial(n,k). Por ejemplo, para determinar el
número de subconjuntos de 2 elementos que se pueden armar a partir de un conjunto de 4 elementos,
evaluamos binomial(4,2) y el resultado es 6.
Se cuenta también con un comando para evaluar los factoriales: factorial(). Como argumento se debe
colocar el número del cual queremos se calcule el factorial.

Es posible también establecer explícitamente los subconjuntos resultantes. Por ejemplo, creamos el
conjunto E de 4 elementos con la entrada

E = [1,2,3,4]

y con el comando S = Subsets(E,2) creamos todos los 6 posibles subconjuntos de 2 elementos que
resultan del conjunto E. Si mostramos S se obtiene

Subsets of {1, 2, 3, 4} of size 2

Con el comando cardinality() se muestra la cardinalidad del conjunto, es decir, el número de ele-
mentos. En este caso, al evaluar S.cardinality() se obtendrá como resultado 6.

Si deseamos ver todos los subconjuntos, usamos el comando list(). En este caso, al evaluar S.list()
se obtendrá la lista

[{1, 2}, {1, 3}, {1, 4}, {2, 3}, {2, 4}, {3, 4}]

que contiene todos los posibles subconjuntos de 2 elementos que se obtuvieron a partir del conjunto E.
Para mostrar un subconjunto aleatorio usamos random_element(). Por ejemplo, evaluando

S.random_element() se obtiene un elemento aleatorio de S. En este caso se obtuvo

{2, 3}

Para acceder a algún subconjunto específico usamos los corchetes. Los subconjuntos se encuentran nume-
rados desde el 0. Por ejemplo, para acceder al primer elemento de S evaluamos S[0] y obtenemos

{1, 2}

2.9. Estadística
2.9.1. Media

Con la función mean() se obtiene la media de una determinada serie de datos. En los argumentos se
debe colocar la lista de valores de los cuales se debe calcular el valor medio. Por ejemplo, para calcular la
media entre el valor de π y de e, evaluamos mean([pi,e]) y el resultado será

1/2*pi + 1/2*e

Si queremos que el resultado con decimales, debemos colocar los elementos de la lista como flotantes, es de-
cir, en el caso anterior, al evaluar mean([float(pi),float(e)]) se obtendrá 2,929937241024419.
Es posible hallar la media de una lista de números aleatorios. Para ello primero creamos la lista. Por ejem-
plo, con el comando

n = [random() for i in range(100)]

se crea una lista de 100 números aleatorios entre 0 y 1. Con el comando list_plot(n) se puede graficar
la lista. El gráfico resultante se muestra en la figura (23).

Figura 23: Lista de 100 números aleatorios entre 0 y 1

Con el comando mean() se puede determinar el valor medio. En este caso, al evaluar mean(n), la
media resulta ser 0,4536961277664366.
Para agregar elementos a la lista se usa el comando append().

2.9.2. Mediana

La mediana representa el valor de la variable de posición central en un conjunto de datos ordenados. Para
hallar la mediana de una lista de datos usamos el comando median(), y en los argumentos se debe colocar
la lista de datos. Por ejemplo, de la lista n de números aleatorios anterior, podemos hallar la mediana.
Evaluando median(n) obtenemos que la mediana es 0,4345872761348995.

2.9.3. Moda

La moda es el valor con una mayor frecuencia en una distribución de datos. Para encontrar la moda
de una lista de datos se usa la función mode(). En los argumentos se debe colocar la lista de datos. Por
ejemplo, creemos una lista de 100 números enteros aleatorios entre 0 y 9. Esto lo realizamos usamos la
función floor() que devuelve la parte entera de un número decimal. La lista la creamos con la entrada

m = [floor(10*random()) for i in range(100)]

El gráfico de la lista m se muestra en la figura (24).

Figura 24: Lista de 100 números enteros aleatorios entre 0 y 9

Encontramos la moda al evaluar mode(m) y el resultado es 9.
La función count() cuenta las veces que aparece un número en una lista. Se puede conocer la frecuencia
de todos los elementos de la lista evaluando la entrada

[m.count(i) for i in range(10)]

y el resultado es

[5, 11, 10, 11, 12, 10, 11, 9, 7, 14]

Es decir, el 0 apareció en la lista 5 veces, el 1 apareció 11 veces y así sucesivamente. Se puede comprobar
que el 9 es la moda.

2.9.4. Media móvil

Las medias móviles son una lista de números en la cual cada uno es el promedio de un subconjunto de
los datos originales. Esto se utiliza para disminuir el ruido en las señales y conseguir gráficas más suaves.

En Sage esto se consigue usando la función moving_average(). Como primer argumento se debe
ingresar la lista de datos originales y como segundo argumento el número de elementos que tendrán los
subconjuntos de los cuales se obtendrá el promedio. El resultado será una lista con un número de elementos
igual al número de elementos de la lista originales menos el número de los elementos de los subconjuntos
más 1.

Por ejemplo, de la lista m creada anteriormente, creemos una lista cuyos elementos sean el promedio
del subconjunto de 10 elementos formado por el elemento de la lista y los siguientes 9 elementos. Para esto
evaluamos moving_average(m,10) y el resultado será una lista con 100− 10 + 1 = 91 elementos.

Evaluemos las siguientes entradas para comprobar como se suaviza la señal:

ma1 = moving_average(m,1)
ma5 = moving_average(m,5)
ma10 = moving_average(m,10)
lp1 = list_plot(ma1,color=’blue’,plotjoined=True,legend_label=’ma1’)
lp5 = list_plot(ma5,color=’black’,plotjoined=True, thickness=2, legend_label=’ma5’)
lp10 = list_plot(ma10,color=’red’,plotjoined=True, thickness=3,legend_label=’ma10’)
lp1+lp5+lp10

Figura 25: Media móvil

Al evaluar los comandos anteriores, resulta el gráfico de la figura (25). La señal de color azul es la de
los datos originales, la de color negro la de la media móvil tomando subconjuntos de 5 elementos y la de
color rojo la de la media móvil tomando subconjuntos de 10 elementos.

2.9.5. Desviación estándar y varianza

La desviación estándar se una serie de datos se puede determinar usando la función std(), en donde,
como argumento, se debe incluir la lista de datos. Por ejemplo, de la lista m creada anteriormente, al evaluar
std(m) se obtiene la desviación estándar. En este caso, en pantalla se muestra

7/30*sqrt(1579/11)

Para obtener el resultado con decimales hay que recordar que se debe evaluar el comando float(std(m)).
En este caso, el resultado es 2,795577315198513. Para determinar la varianza se usa la función variance().
Para la lista m, evaluando float(variance(m)) se obtiene 7,815252525252525.

2.9.6. Histogramas

Se pueden hacer histogramas de una lista de datos con la función histogram(). En los argumentos
se debe colocar la lista de datos y el número de bins. Por ejemplo, de la lista m creada, con el comando
histogram(m,bins=5) hacemos un histograma con 5 bins de la lista. El gráfico resultante se muestra
en la figura (26).

Figura 26: Histograma

2.9.7. Funciones de ajuste

En Sage se pueden hacer ajustes de los datos con funciones con parámetros arbitrarios. Sage determina
el valor de los parámetros que minimizan el error entre los valores ajustados y los valores en los datos. Para
esto se usa la función find_fit(), en la cual, en sus argumentos se debe colocar los datos a ajustar y la
función con los parámetros arbitrarios con la cual se quiere realizar el ajuste. Procedemos con un ejemplo.

Creemos los datos. Para esto debemos tener una lista con valores para x y para y. En este caso, los
valores de x serán los números del 0 al 90 y los valores para y los elementos de la lista ma10. Esto lo
hacemos con la entrada

data = zip(range(91),ma10)

Después declaramos los parámetros arbitrarios de la función de ajuste. En este caso declararemos los pará-
metros A, B, C, D, E. También se debe declarar la variable x de la función. Esto se hace evaluando la
entrada

var(’A,B,C,D,E,x’)

Después se debe indicar la función a ajustar. En este caso se ajustarán los datos con la función f(x) =
Ax4 +Bx3 + Cx2 +Dx+ E. Evaluamos la entrada

f(x) = A*x^4+B*x^3+C*x^2+D*x+E

Para realizar el ajuste evaluamos

fit = find_fit(data,f,solution_dict=True)

donde data son los datos a ajustar y f es la función de ajuste. Para mostrar el ajuste debemos reemplazar
los valores hallados para los parámetros en la función de ajuste. Esto lo hacemos evaluando

show(f.subs(fit))

y el resultado será

x 7→ − (9,79076766096× 10−07) x4 + 0,000162544567578x3

−0,00827108952225x2 + 0,135685988739x+ 4,26126699875

que es la función que minimiza el error.
Para graficar la función de ajuste y los datos, evaluamos

plot(f.subs(fit),(x,0,90),color=’blue’,thickness=3)+lp10

Figura 27: Ajuste f(x) = Ax4 +Bx3 + Cx2 +Dx+ E

Figura 28: Ajuste f(x) = (A1x2 +B1x+ C1) eD1x

donde lp10 corresponde a la gráfica de la lista ma10. El gráfico de la lista con el ajuste f(x) = Ax4 +
Bx3 + Cx2 +Dx+ E se muestra en la figura (27).

Se pueden hacer ajustes con otras funciones. Por ejemplo, para hacer un ajuste con una función f(x) =
(A1x2 +B1x+ C1) eD1x, evaluamos las siguientes entradas

data = zip(range(91),ma10)
var(’A,B,C,x’)
f(x) = (A1*x^2+B1*x+C1)*exp(D1*x)
fit = find_fit(data,f,solution_dict=True)
show(f.subs(fit))

x 7→ ((2,07040660837× 10−10)x2

− (3,61670061121× 10−08) x
+1,58014300478× 10−06)e(0,250671948791x)

Para graficar el ajuste y los datos, evaluamos

plot(f.subs(fit),(x,0,90),color=’blue’,thickness=3)+lp10

El gráfico de la lista con el ajuste f(x) = (A1x2 + B1x + C1) eD1x se muestra en la figura (28). En la
figura, el ajuste no es tan correcto pero se puede notar que Sage puede hacer ajustes con cualquier función
y no sólo con polinomios, y depende de nosotros escoger la función que mejor ajusta los datos.

3. SageManifolds
SageManifolds es un paquete de Sage que implementa geometría diferencial y cálculo tensorial. Opera

con variedades diferenciales reales en cualquier dimensión.
En SageManifolds se pueden definir varios mapas para una sola variedad y definir campos tensoriales, cuyas
representaciones en cada mapa son calculadas por el programa.

3.1. Definición de una variedad
Una variedad es un objeto que localmente es parecido a Rn. Para definirlo en Sage usamos Manifold().

En los argumentos se debe colocar un número entero, el cual es la dimensión de la variedad, se debe colocar
el nombre de la variedad entre comillas simples.
Si queremos que un símbolo de LATEX represente a la variedad colocamos en los argumentos
latex_name=r’ ’, entre las comillas simples se debe colocar el símbolo que representará la variedad.

Por defecto, los elementos de la base de la variedad se numeran desde el cero. Si queremos que empiecen
desde otro número, en los argumentos debemos incluir start_index= y colocar el número desde el cual
empezará la numeración.
Por ejemplo, definamos una variedad E2 de 2 dimensiones, cuyos elementos de la base se numeren desde
el 1. Esto lo hacemos al evaluar

E2 = Manifold(2, ’E^2’, latex_name=r’\mathbb{E}^2’, start_index=1)

Con esta entrada se crea la variedad de 2 dimensiones, llamada E^2, que será representada por E2 y la
numeracíon de los índices empieza en 1.

Si evaluamos print(E2) se mostrará

2-dimensional manifold ’E^2’

mientras que si evaluamos show(E2) se mostrará

E2

que es el símbolo que representa la variedad.

3.2. Sistema de coordenadas
Para comprender como se define un sistema de coordenadas en SageManifolds procedemos con un

ejemplo. En la variedad E2 definamos coordenadas cartesianas. Para esto evaluamos la siguiente entrada

cart.<x,y> = E2.chart()

En este caso, el objeto que tiene las coordenadas se llama cart. Con el método .<x,y> se indican como
se llaman las coordenadas. En este caso se necesitan 2 coordenadas x y y. Por otro lado, para indicar que el
objeto cart.<x,y> corresponde a las coordenadas de la variedad, asociamos el objeto a E2.chart().

Al evaluar show(cart) se obtendrá (
E2, (x, y)

)
donde se indica la variedad y el sistema de coordenadas definido. En los argumentos de chart() se debe
incluir los rangos de las coordenadas. Por defecto, el rango de las coordenadas es [−∞,+∞]. En este caso
no colocamos el rango porque el rango de las coordenadas cartesianas es el rango predeterminado.

El objeto cart es una lista. Para acceder, por ejemplo a x y recordando que la numeración empie-
za en 1, evaluamos cart[1]. Podemos verificar que el segundo elemento del objeto es y al evaluar
y is cart[2] y se mostrará en pantalla

True

dado que es verdad.
Para hacer un gráfico de las coordenadas en funcíon de si mismas usamos la función plot(). En este

caso, al evaluar cart.plot() se obtiene el gráfico de la figura (29). En el gráfico, las líneas horizontales

-5 5
x

-5

5

y

Figura 29: Coordenadas cartesianas en función de sí mismas

corresponden a y constante y las líneas verticales a x constante.
Se pueden definir más sistemas de coordenadas en la variedad. Para esto, antes veamos como crear

subconjuntos de la variedad.
Con la función open_subset() se crean subconjuntos abiertos de la variedad. En los argumentos se
debe colocar el nombre del subconjunto entre comillas simples. Por ejemplo, creemos los subconjuntos
abiertos P y H. Esto lo hacemos evaluando P = E2.open_subset(’P’) y también
H = E2.open_subset(’H’). Si evaluamos print H se obtendrá

open subset ’H’ of the 2-dimensional manifold ’E^2’

Ahora definamos otros sistemas coordenados en los subconjuntos creados. En el subconjunto P definamos
coordenadas polares (r, φ) donde r ≥ 0 y 0 ≤ φ ≤ φ. Para eso evaluamos

pol.<r,phi> = P.chart(r’r:(0,oo):r phi:(0,pi):\phi’)

En H se puede definir otro sistema de coordenadas. Por ejemplo, definamos coordenadas hiperbólicas (u, v)
donde v ≥ 0 y u está definido en todos los reales. Esto lo realizamos al evaluar

hip.<u,v> = H.chart(r’u v:(0,oo)’)

Con la función atlas() se pueden ver todos los sistemas de coordenadas definidos en la variedad. Por
ejemplo, si evaluamos show(E2.atlas()) se mostrará[(

E2, (x, y)
)
, (P, (r, φ)) , (H, (u, v))

]
donde se pueden ver los sistemas coordenados definidos y los conjuntos en los cuales están definidos.

3.3. Funciones de transición
Las funciones de transición nos permiten pasar de un sistema de coordenadas hacia otro sistema de

coordenadas dentro de la misma variedad. Por ejemplo, nos permiten pasar de coordenadas cartesianas a
polares. En Sage usamos la función transition_map() para definir funciones de transición. En los
argumentos colocamos el sistema de coordenadas, las funciones de transformación y las restricciones en las
coordenadas.

Por ejemplo, en los casos anteriores, sabemos que existen las siguientes relaciones entre las coordenadas
cartesianas y las polares

r =
√
x2 + y2

φ = arctan(y/x)

En Sage definimos estas relaciones al evaluar

cart_to_pol=cart.transition_map(pol,(sqrt(x^2+y^2),arctan(y/x)),
restrictions1=x>0)

donde el objeto cart_to_pol contiene las funciones de transformación desde coordenadas cartesianas a
polares. En los argumentos de transition_map() se tiene que pol es el sistema coordenado al cual
se llega después de aplicadas las funciones de transición, (sqrt(x^2+y^2), arctan(y/x)) son la
funciones de como se transforma la primera y la segunda coordenadas del sistema polar, y
restrictions1= x>0 son las restricciones, en este caso, se restringe los valores de la coordenada x.

Al evaluar print(cart_to_pol) se mostrará

coordinate change from chart (P,(x, y)) to chart (P, (r,phi))

mientras que si evaluamos show(cart_to_pol.display()) se mostrará{
r =

√
x2 + y2

φ = arctan
(
y
x

)
Si queremos establecer las funciones de transición inversas podemos usar la función inverse().

En algunas ocasiones no podrá determinar la transformación inversa y se mostrará un mensaje de error.
Por ejemplo, si para hallar las funciones de transformación inversas, es decir, las transformaciones desde
coordenadas polares a cartesianas, evaluamos cart_to_pol.inverse() se mostrará

Traceback (click to the left of this block for traceback)
...
RuntimeError: The system could not be solved; use set_inverse() to set
the inverse manually.

Si el programa no puede definir la transformación inversa, la podemos definir manualmente con la función
set_inverse(), en donde se deben colocar como argumentos las funciones de transformación. Por
ejemplo, sabemos que

x = r cos(φ)
y = r sin(φ)

Definimos esta transformación al evaluar

cart_to_pol.set_inverse(r*cos(phi), r*sin(phi))

dado que permite el paso de coordendas polares a cartesianas. Al evaluar el comando anterior se mostrará
en pantalla

Check of the inverse coordinate transformation:
x == x
y == y
r == r
phi == arctan(sin(phi)/cos(phi))

Observatorio Astronómico de Quito-Escuela Politécnica Nacional 52



3.4 Coordenadas en términos de otras coordenadas Manual de cálculo científico en SageMath

es decir, Sage verifica si las transformaciones están definidas correctamente.
Se puede realizar los mismo con las coordenadas hiperbólicas. En este caso tenemos que{

u = −1

2
ln
(y
x

)
v =
√
xy

Definimos las funciones de transición para el paso de coordenadas cartesianas a coordenadas hiperbólicas
al evaluar

cart_hip=cart.transition_map(hip,(-log(y/x)/2,sqrt(x*y)),
restrictions1=x>0, restrictions2=y>0)

y dado que {
x = veu

y = ve−u

definimos la trasformación inversa evaluando

cart_hip.set_inverse(v*e^u, v*e^-u)

y se mostrará

Check of the inverse coordinate transformation:
x == x
y == y
u == u
v == v

indicando que las transformaciones están bien definidas entre coordenadas cartesianas e hiperbólicas.
Si evaluamos show(E2.atlas()) se mostrará[(

E2, (x, y)
)
, (P, (r, φ)) , (H, (u, v)) , (P, (x, y)) , (H, (x, y))

]
es decir, el atlas tiene más componentes dado que, en los subconjuntos P y H de la variedad se definen las
coordenadas cartesianas por medio de las funciones de transición.

3.4. Coordenadas en términos de otras coordenadas
Si se grafican las coordenadas polares en términos de sí mismas, al evaluar pol.plot() se obtendrá

el gráfico de la figura (30), en el cual, las rectas horizontales son a φ constante y las verticales a r constante.

Figura 30: Coordenadas polares en función de sí mismas

Pero si queremos graficar las coordenadas polares en función de las cartesianas debemos especificarlo
en los argumentos de plot(). En este caso, al evaluar pol.plot(cart), donde se especifica en los
argumentos las coordenadas cartesianas, se obtiene el gráfico de las coordenadas polares en función de las
cartesianas que se muestra en la figura (31).

Figura 31: Coordenadas polares en función de las cartesianas

Figura 32: Coordenadas hiperbólicas en función de las cartesianas

También podemos graficar las coordenadas hiperbólicas en función de las cartesianas al evaluar
hip.plot(cart, ranges={u:[-1,1], v:[0,1]})
donde se puede notar que en los argumentos se incluye el rango de las coordenadas hiperbólicas. El gráfico
se muestra en la figura (32).

Estos gráficos se pueden hacer dado que las coordenadas hiperbólicas y polares están definidas en sub-
conjuntos de la variedad, que es en donde están definidas las coordenadas cartesianas. Pero si queremos
hacer un gráfico de coordenadas cartesianadas en función de las coordenadas hiperbólicas no se los puede
realizar. Si evaluamos cart.plot(hip) se mostrará

Traceback (click to the left of this block for traceback)
...
TypeError: the domain of chart (E^2, (x, y)) is not included in that of
chart (H, (u, v))

Para realizar este gráfico se deben restringir las coordenadas definidas en la variedad al subconjunto. Por
ejemplo, para graficar las coordenadas cartesianas, que están definidas en la variedad, en función de las
coordenadas hiperbólicas definidas en H , se deben restringir las coordenadas cartesianas a H . Esto se lo
realiza con la función restrict(). En este caso, evaluamos cartH = cart.restrict(H).

Para hacer el gráfico evaluamos

cartH.plot(hip, ranges={x:[0.02,1], y:[0.02,1]})

y se obtiene el gráfico de la figura (33).

Figura 33: Coordenadas cartesianas en función de las hiperbólicas en H

Lo mismo se puede realizar en función de las coordenadas polares. Primero debemos restringir las
coordenadas cartesiana a P al evaluar
cartP = cart.restrict(P)
y para el gráfico al evaluar
cartP.plot(pol)
Ver la figura (34).

Figura 34: Coordenadas cartesianas en función de las polares en P

3.5. Puntos en la variedad
Para explicar como se definen puntos en la variedad procedemos con un ejemplo.

Para definir el punto a = (0, 0) de la variedad E2 en coordenadas cartesianas evaluamos

a = E2((0,0), chart=cart, name=’a’)

donde se definen las coordenadas del punto, el sistema coordenado (chart=) y el nombre del punto entre
comillas simples (name=’’). Definamos el punto b = (1, 1) en coordenadas cartesianas evaluando

b = E2((1,1), chart=cart, name=’b’)

Si se evalúa a in E2, a in H, a in P se obtendrá

(True, False, False)

es decir, el punto a pertenece a la variedad pero no a los subconjuntos P y H . En cambio, al evaluar
b in E2, b in H, b in P se obtendrá

(True, True, True)

es decir, el punto b pertenece tanto a la variedad como a los subconjuntos P y H .
Para mostrar las coordenadas de un punto usamos coord(). En los argumentos se debe incluir el

sistema coordenado en el cual queremos que se muestren las coordenadas del punto. Por ejemplo, al evaluar
b.coord(cart) se obtiene

(1, 1)

que son las coordenadas cartesianas del punto b. Si evaluamos b.coord(pol) se obtiene

(sqrt(2), 1/4*pi)

que son las coordenadas polares del punto b. Y si evaluamos b.coord(hip) se obtendrá

(0, 1)

que son las coordenadas hiperbólicas del punto b.

3.6. Espacio tangente en un punto
El espacio tangente en un punto es el espacio vectorial formado por todos los vectores tangentes al punto

de la variedad. En Sage, los espacios tangentes a un punto se declaran con la función tangent_space().
Por ejemplo, el espacio tangente del punto a se lo declara con T_a = a.tangent_space(). Si eva-
luamos print T_a se mostrará

tangent space at point ’a’ on 2-dimensional manifold ’E^2’

La dimensión del espacio tangente es la misma que la dimensión de la variedad. Se puede comprobar esto
al evaluar dim(T_a). En este caso, el resultado es 2.

Al espacio tangente se encuentran asociadas bases que lo generan. Para mostrar dichas bases usa-
mos la función bases(). Por ejemplo, la base del espacio tangente del punto a se la muestra al evaluar
show(T_a.bases()) y se muestra [(

∂

∂x
,
∂

∂y

)]
Al evaluar

T_b = b.tangent_space()
show(T_b.bases())

estamos declarando el espacio tangente en b y mostramos la base del espacio. El resultado es[(
∂

∂x
,
∂

∂y

)
,

(
∂

∂r
,
∂

∂φ

)
,

(
∂

∂u
,
∂

∂v

)]
dado que b está definido tanto en la variedad como en los subconjuntos P y H donde están definidas las
coordenadas polares e hiperbólicas respectivamente. En este caso, para conocer la base predeterminada
usamos la función default_basis(). Si evaluamos show(T_b.default_basis()) se mostrará(

∂

∂x
,
∂

∂y

)
es decir, la base predeterminada del espacio tangente en b es la base cartesiana.

3.7. Vectores en el espacio tangente
Para explicar como se definen vectores en el espacio tangente procedemos con un ejemplo. Declaremos

el vector de componentes V = (−3, 2) en el espacio tangente de a. Para esto evaluamos

V = T_a((-3, 2), name=’V’)

donde se incluyen las componentes del vector y su nombre entre comillas simples (name=’’). Si se evalúa
print V se mostrará

tangent vector V at point ’a’ on 2-dimensional manifold ’E^2’

Para mostrar al vector con sus componentes se puede evaluar show(V.disp()) y se muestra

V = −3
∂

∂x
+ 2

∂

∂y

En el espacio tangente en b declaremos el vector W = (−3, 2) evaluando

W = T_b((-3, 2), name=’W’)

Si intentamos sumar V +W se mostrará un error debido a que los vectores pertenecen a espacios vectoriales
diferentes y no se los puede sumar de esa forma. Evaluando V+W da como resultado

Traceback (click to the left of this block for traceback)
...
TypeError: unsupported operand parent(s) for ’+’: ’tangent space at
point ’a’ on 2-dimensional manifold ’E^2’’ and ’tangent space at point
’b’ on 2-dimensional manifold ’E^2’’

Para conocer las componentes de un vector en cualquier base del espacio procedemos como se indicará
a continuación. Con la función display() se muestran las componentes del vector. En sus argumentos
se debe incluir la base en la cual se quiere mostrar el vector. Por ejemplo, el espacio tangente en b tiene 3
bases. Mostremos el vector W = (−3, 2) en las tres bases. En la base cartesiana, evaluando
show(W.display(T_b.bases()[0]))
se obtiene

W = −3
∂

∂x
+ 2

∂

∂y

En este caso es suficiente con evaluar show(W.display()) dado que la base predeterminada es la
cartesiana. Para mostrar el vectorW en la base polar evaluamos show(W.display(T_b.bases()[1])),
dado que en la lista de bases del espacio tangente, la base polar es la segunda base asociada. Se mostrará

W = −1

2

√
2
∂

∂r
+

5

2

∂

∂φ

Por último, si queremos conocer las componentes de W en la base hiperbólica evaluamos
show(W.display(T_b.bases()[2]))
y se muestra

W = −5

2

∂

∂u
− 1

2

∂

∂v

3.8. Funciones en la variedad
Las funciones son campos escalares en la variedad. En Sage se los declara con las función

scalar_field(). En los argumentos se debe incluir entre llaves el sistema de coordenadas que se usará
para definir la función y el campo escalar en función de las coordenadas, además se debe incluir el nombre
entre comillas simples. Por ejemplo, definamos el campo escalar f(x, y) = x2 + 2y2 en la variedad. Para
esto evaluamos

f = E2.scalar_field({cart: x^2+2*y^2}, name=’f’)

Si evaluamos show(f.display()) se mostrará

f : E2 −→ R
(x, y) 7−→ x2 + 2 y2

on P : (r, φ) 7−→
(
sin (φ)2 + 1

)
r2

on H : (u, v) 7−→ v2
(
e(4u) + 2

)
e(−2u)

Si bien la función está definida en coordenadas cartesianas, Sage muestra también su definición en los
sistemas coordenados definidos sobre los subconjuntos de la variedad o sobre la variedad misma.

Para conocer el álgebra a la cual pertenece la función usamos parent(). Por ejemplo, si evaluamos
print f.parent() se obtiene

algebra of scalar fields on the 2-dimensional manifold ’E^2’

Para evaluar la función en algún punto de la variedad procedemos de la siguiente manera. Por ejemplo, si
queremos evaluar la función en el punto a, lo hacemos con la entrada f(a) y se muestra

0

Se pueden definir más funciones que sean función del campo escalar definido. Por ejemplo, definamos la
función g = f 2−2f . Esto lo hacemos al evaluar g = f*f - 2*f. Si evaluamos show(g.display())
se obtiene

E2 −→ R
(x, y) 7−→ x4 + 4 y4 + 4 (x2 − 1)y2 − 2x2

on P : (r, φ) 7−→
(
sin (φ)4 + 2 sin (φ)2 + 1

)
r4 − 2

(
sin (φ)2 + 1

)
r2

on H : (u, v) 7−→
(
v4
(
e(8u) + 4 e(4u) + 4

)
− 2 v2

(
e(6u) + 2 e(2u)

))
e(−4u)

Si queremos conocer la forma de la función en un determinado sistema de coordenadas, debemos espe-
cificarlo dentro de los argumentos de la función display(). Por ejemplo, si nos interesa la forma de g en
coordenadas hiperbólicas evaluamos show(g.display(hip)) y se obtiene

E2 −→ R
on H : (u, v) 7−→

(
v4
(
e(8u) + 4 e(4u) + 4

)
− 2 v2

(
e(6u) + 2 e(2u)

))
e(−4u)

3.9. Módulo de campos vectoriales
El módulo de campos vectoriales se define con la función vector_field_module(). Al evaluar

XE = E2.vector_field_module()
se crea el módulo de campos vectoriales en la variedad E2.

Para mostrar las bases usamos la función print_bases(). En este caso, evaluamos
XE.print_bases() se obtiene

Bases defined on the free module X(E^2) of vector fields on the
2-dimensional manifold ’E^2’:
- (E^2, (d/dx,d/dy)) (default basis)

donde se indica la base predeterminada. Si queremos mostrar la base por defecto podemos usar
default_basis().
Si evaluamos show(XE.default_basis()) se obtendrá(

E2,

(
∂

∂x
,
∂

∂y

))
y es la única base que genera el modulo de campos vectoriales en la variedad. Si realizamos lo mismo pero
en el subconjunto P , es decir, si evaluamos

XP = P.vector_field_module()
XP.print_bases()

se obtendrá

Bases defined on the free module X(P) of vector fields on the open
subset ’P’ of the 2-dimensional manifold ’E^2’:
- (P, (d/dr,d/dphi)) (default basis)
- (P, (d/dx,d/dy))

Si evaluamos

e = XE.default_basis()
print(e)

se obtiene

coordinate frame (E^2, (d/dx,d/dy))

Si queremos acceder al primer elemento de la base predeterminada entonces evaluamos show(e[1]) y
se obtiene

∂

∂x

Se pueden declarar campos vectoriales en la variedad a partir de las bases del módulo de campos vecto-
riales. Esto se lo realiza con la función vector_field(). En sus argumentos se debe incluir el nombre
del campo vectorial entre comillas simples precedido de name=. Por ejemplo, definamos el campo vectorial
vE evaluando la entrada vE = E2.vector_field(name=’vE’). Para indicar las componentes del
campo vectorial necesitamos una base. En este caso, evaluando

vE[e,:] = [1, -2]

que indica que las componentes del campo vectorial vE en la base predeterminada del módulo de campos
vectoriales son (1,−2). Si evaluamos
show(vE.display(e))
se tendrá

vE =
∂

∂x
− 2

∂

∂y

Para graficar el campo vectorial podemos evaluar
vE.plot(chart=cart,max_value=2, nb_values=4, scale=0.5, aspect_ratio=1)
donde en los argumentos se incluye el sistema coordenado a considerarse (chart=), se incluye el máximo
valor en el eje vertical (max_value=), el número de vectores a graficarse horizontalmente (nb_values=),
la escala de los vectores (scale=) y la relación entre las escalas de los ejes (aspect_ratio=). El gráfico
se muestra en la figura (35).

Se puede restringir un campo vectorial a un subconjunto de la variedad con la función restrict().
Por ejemplo con vP = vE.restrict(P) restringimos el campo vectorial vE al subconjunto P . Con
esto podemos obtener el campo vectorial en la base que se ha definido en algún subconjunto mas no en la
variedad. Por ejemplo, al evaluar show(vP.display()) se obtiene

vE = (cos (φ)− 2 sin (φ))
∂

∂r
+

(
−2 cos (φ) + sin (φ)

r

)
∂

∂φ

que es el campo vectorial en la base polar definida en P . Podemos graficar el campo vectorial en coordena-
das polares evaluando

Figura 35: Campo vectorial vE = ∂
∂x
− 2 ∂

∂y

Figura 36: Campo vectorial (cos (φ)− 2 sin (φ)) ∂
∂r

+
(
−2 cos(φ)+sin(φ)

r

)
∂
∂φ

vP.plot(chart=pol,ranges={r:[0.1,1], phi:[0.1,3.1]},max_value=4,
nb_values=5,scale=0.2)

donde se han definido rangos para las coordenadas polares. Ver el resultado en la figura (36).
Se pueden multiplicar campos escalares por campos vectoriales para establecer nuevos campos vectoria-

les. Por ejemplo, declaremos el campo vectorial xE = 2∗vE−f ∗vE con la entrada xE = 2*vE-f*vE.
Al evaluar show(xE.display()) se obtendrá el campo vectorial resultante(

−x2 − 2 y2 + 2
) ∂

∂x
+
(
2x2 + 4 y2 − 4

) ∂
∂y

3.10. Campos vectoriales actuando en funciones
Cuando un campo vectorial actúa en una función se obtiene un campo escalar. Para esto procedemos

con un ejemplo. Si queremos que el campo vectorial vE actúe sobre la función f , entonces evaluamos
vEf = vE(f). vEf será un nuevo campo escalar. Si evaluamos show(vEf.display()) se mostrará

vE (f) : E2 −→ R
(x, y) 7−→ 2x− 8 y

on P : (r, φ) 7−→ 2 r(cos (φ)− 4 sin (φ))
on H : (u, v) 7−→ 2 v

(
e(2u) − 4

)
e(−u)

Los campos escalares pueden actuar sobre puntos de la variedad. Por ejemplo, al evaluar vEf(b) se
evalúa el campo vEf en el punto b. Si se evalúa vEf(b) se obtendrá -6 en este ejemplo.

3.11. Métrica de la variedad
El tensor métrico de la variedad es un tensor de rango 2 simétrico que permite medir distancias en una

variedad.
Una variedad Riemanniana es una variedad que tiene una métrica definida positiva. En Sage, una métrica
Riemanniana se declara con la función riemann_metric() y en los argumentos debe incluirse el nom-
bre de la variedad entre comillas simples. Por ejemplo, asumamos que la variedad E2 es Riemanniana y
definamos su tensor métrico h evaluando

h = E2.riemann_metric(’h’)

con esto se crea el objeto h que es el tensor métrico de la variedad.
Las variedades Lorentzianas son variedades de dimensión n que tienen asociadas un tensor métrico con

signatura (1, n−1). En Sage, este tipo de tensores métricos se crean con la función lorentz_metric().
El ingreso de las componentes del tensor métrico se lo realiza componente a componente, y se deben

especificar las componentes no nulas del tensor métrico en la base predeterminada. La identificación de las
componentes se la realiza por medio de sus índices. Por ejemplo, supongamos que el tensor métrico h de la
variedad E2 en forma matricial en coordenadas cartesianas es

h =

(
1 0
0 1

)
es decir, las componentes no nulas del tensor h son h11 = 1 y h22 = 1. En Sage, ingresamos

h[1,1], h[2,2] = 1, 1

y no se deben indicar las demás componentes dado que Sage les asocia el valor de 0.
Para mostrar el tensor métrico en la base predeterminada se tienen 2 opciones: con show(h.display())

se obtiene
h = dx⊗ dx+ dy ⊗ dy

y con show(h[:]) se obtiene (
1 0
0 1

)
El tensor métrico debe ser simétrico. Para verificar que el tensor que hallamos indicado cumple con esta

propiedad se puede evaluar h.symmetries() y se obtiene

symmetry: (0, 1); no antisymmetry

Por otro lado, la norma al cuadrado | v |2 de un campo vectorial v está definida por

| v |2= hijv
ivj

donde se tiene una sumatoria en i y en j usando el convenio de suma. En Sage procedemos como en el
siguiente ejemplo. La norma del campo vectorial vE se puede calcular evaluando

vE^2=h(vE,vE)

Para mostrar la respuesta, evaluamos show(h(vE,vE).disp()) y se obtiene

h (vE, vE) : E2 −→ R
(x, y) 7−→ 5

on P : (r, φ) 7−→ 5
on H : (u, v) 7−→ 5

entonces su norma es
√

5.
El producto interno euclídeo < v,w > entre dos campos vectoriales v y w está definido por

< v,w >= hijv
iwj

Para ejemplificarlo, declaremos el campo vectorial wE = y2 ∂
∂x
− x ∂

∂y
evaluando

Observatorio Astronómico de Quito-Escuela Politécnica Nacional 61



3.12 Derivada y conexión Manual de cálculo científico en SageMath

wE = E2.vector_field(name=’wE’)
wE[e,:] = [y^2, -x]

El producto interno entre vE y wE se lo puede calcular evaluando

show(h(vE,wE).disp())

y se obtendrá
h (vE,wE) : E2 −→ R

(x, y) 7−→ y2 + 2x

on P : (r, φ) 7−→ r2 sin (φ)2 + 2 r cos (φ)
on H : (u, v) 7−→

(
v2 + 2 ve(3u)

)
e(−2u)

Para mostrar el tensor métrico en las demás bases se debe especificar la base. Por ejemplo, para mostrar
el tensor métrico en coordenadas polares, evaluamos

show(h.display(pol.frame()))

y se obtiene
h = dr ⊗ dr + r2dφ⊗ dφ

y si deseamos conocer sus componentes en coordenadas hiperbólicas, evaluamos

show(h.display(hip.frame()))

y se obtiene

h = v2
(
e4u + 1

)
e−2udu⊗du+v

(
e4u − 1

)
e−2udu⊗dv+v

(
e4u − 1

)
e−2udv⊗du+

(
e4u + 1

)
e−2udv⊗dv

3.12. Derivada y conexión
En una variedad se define el operador derivada covariante∇µ, que al aplicarlo a un tensor Aβα se obtiene

∇µA
β
α =

∂Aβα
∂xµ

+ Γβ µγA
γ
α − Γγ αµA

β
γ

que cumple con la ley de transformación covariante y donde Γβ µγ son los símbolos de Christoffel de
segunda clase que en función del tensor métrico están definidos por

Γβ µγ =
1

2
gβλ(∂γgλµ + ∂µgλγ − ∂λgµγ)

En Sage, tanto el operador derivada covariante y los símbolos de Christoffel de segunda clase, se los
calcula con la función connection(). Por ejemplo, al evaluar

nab = h.connection()

estamos definiendo el operador derivada covariante asociado a la métrica h y los símbolos de Christoffel de
segunda clase. Si evaluamos print nab se obtiene

Levi-Civita connection ’nabla_h’ associated with the Riemannian metric
’h’ on the 2-dimensional manifold ’E^2’

y si evaluamos

show(nab)

se obtiene
∇h

Para determinar la derivada covariante de algún objeto, se debe aplicar el operador al objeto. Por ejem-
plo, para hallar la derivada covariante del campo vectorial vE, evaluamos

show(nab(vE).disp())

y se obtiene
∇hvE = 0

y si lo hacemos para el campo vectorial wE evaluando

show(nab(wE).disp())

se obtiene
wE =

∂

∂x
⊗ dy − ∂

∂y
⊗ dx

Para mostrar el resultado en otra base, debemos especificarla. Por ejemplo, evaluando

show(nab(wE).disp(pol.frame()))

se obtiene
wE = r

∂

∂r
⊗ dφ− 1

r

∂

∂φ
⊗ dr

3.13. Tensor de curvatura
El tensor de Riemann Rα

λµν es el tensor que determina si una variedad es plana o curva. En función del
tensor métrico, sus componentes están dadas por

Rα
λµν = ∂µΓα λν − ∂νΓα λµ + Γβ λνΓ

α
βµ − Γβ λµΓα βν

En Sage, sus componentes se calculan con la función riemann(). Por ejemplo, de la variedad anterior, se
tiene que el tensor de Riemann asociado con h se calcula al evaluar

Riem = h.riemann()

y si evaluamos show(Riem.display()) se obtiene

Riem (h) = 0

Esto indica que la variedad es plana.
El tensor de Ricci Rνρ se lo obtiene de la contracción del tensor de Riemann, es decir,

Rνρ = Rµ
νµρ (1)

En Sage se lo calcula con la función ricci(). Por ejemplo, el tensor de Ricci asociado al tensor métrico
h se lo calcula evaluando

Ric=h.ricci()

y si evaluamos show(Ric[:]), se obtiene (
0 0
0 0

)
La traza del tensor de Ricci recibe el nombre de escalar de curvatura R, es decir,

R = hνρRνρ = Rν
ν

En Sage se lo calcula con la función ricci_scalar(). Siguiendo con el ejemplo, si evaluamos

R=h.ricci_scalar()

estaremos calculando el escalar de curvatura asociado a la métrica h de la variedad. Si evaluamos
show(R.display()) se obtiene

r (h) : E2 −→ R
(x, y) 7−→ 0

on P : (r, φ) 7−→ 0
on H : (u, v) 7−→ 0

Estos tensores son importantes en Relatividad General dado que a partir de ellos se establecen las ecua-
ciones de campo de Einstein para el campo gravitatorio.

